
บทที่   Å
กรณีศึกษาระบบงานขาย และสินคาคงคลัง

จดุประสงคเชิงพฤติกรรม
1. เขาใจระบบขาย และสินคาคงคลังในมุมมองของ Programmer มากขึ้น
2. เขาใจการสรางระบบงานคอมพิวเตอร โดยพิจารณาจากความตองการของผูใช
3. มปีระสบการณ และมองเห็นแนวการออกแบบเมนูทั้งระบบงาน
4. เขาใจแนวการเขียน และการแปลโปรแกรมในอีกรูปแบบหนึ่ง
5. เขาใจการท ํารายงาน และการท ําซํ้ าจากกรณีศึกษานี้มากขึ้น

หวัขอในบทเรียน
6.1  ศึกษาความตองการ
6.2  โครงสรางขอมูล
6.3  วิเคราะหระบบงานเบื้องตน
6.4  โปรแกรมในระบบงานขาย และสินคาคงคลัง

6.4.1  โปรแกรมตาง ๆ ใน MENU.PRG   : สารบัญของโปรแกรม
6.4.2  โปรแกรมตาง ๆ ใน SALE1.PRG  : รวมการขายหลายรูปแบบ
6.4.3  โปรแกรมตาง ๆ ใน SALE2.PRG  : รวมโปรแกรมดานการปรับปรุงขอมูล
6.4.4  โปรแกรมตาง ๆ ใน SALE3.PRG  : รวมโปรแกรมรายงานยอดขาย
6.4.5  โปรแกรมตาง ๆ ใน SALE4.PRG  : รวมโปรแกรมดานสินคาคงคลัง
6.4.6  โปรแกรมตาง ๆ ใน SALE5.PRG  : รวมรายงานอื่น ๆ ที่เกิดขึ้น

ตวัอยางโปรแกรมเกี่ยวกับระบบงานขาย และสินคาคงคลัง  37  ตัวอยาง

หมายเหตุ
ตัวอยางนี้อาจขัดกับความเปนจริงอยูมาก เพราะโปรแกรมขาย หรือสินคาคงคลังนั้น จะมีความซับซอน

อยางมาก ถงึมากทีสุ่ด  กรณีศึกษานี้เกิดจากประสบการณในการเปนลูกคาเปนหลักใหญ  ซึ่งตองการสื่อใหเห็นแนว
การเขยีนโปรแกรมดานการปรับปรุง การทํ ารายงานใหมีความหลากหลาย และซับซอนที่แตกตางกันไป .. หากจะนํ า
ไปใชจริง ตองศึกษาความตองการของผูที่จะนํ าไปใช และพัฒนาตออีกมาก และโปรแกรมทั้งหมดในหนังสือ นัก
ศกึษาสามารถ Download ไปศกึษา ประกอบการเรียนภาคปฏิบัติ

บทที่ 6   กรณีศึกษาระบบงานขาย และสินคาคงคลัง
เนือ่งจากสงัคมทกุวันนี ้ เปนสังคมของนายทุน มีการซื้อขายแลกเปลี่ยนในสังคมตลอดเวลา  ธุรกิจที่หวังผล

ก ําไรจงึอยูไดดวยการขาย และการจัดเก็บสินคาไวในคลัง เพื่อใหการขายไดเงินมาอยางไมตกหลน และการจัดเก็บ
สนิคามปีระสทิธภิาพไมมากไมนอยเกินควร นักการคายุคใหมจึงนํ าคอมพิวเตอรมาใชในหนวยงานกันทั่วหนา

โปรแกรมคอมพิวเตอรที่มีขายอยูในทองตลาด ถูกเขียนขึ้นใหคลอบคลุมกับงานไดในหลาย ๆ ลักษณะมี
ความยดืหยุนสงู  แตมีบางบริษัทที่มีรูปแบบ และความตองการเฉพาะของตน จึงไมอาจนํ าโปรแกรมที่มีขายในทอง
ตลาดมาใชไดอยางเต็มที ่   บริษัทเหลานั้นมักจางนักเขียนโปรแกรม มาพัฒนาระบบคอมพิวเตอรใหตน ตามความ
ตองการที่ไมจ ํากัด

&  6.1 ศึกษาความตองการ
ผูเขยีนไมไดเปนพอคาโดยอาชีพ  จึงศึกษาระบบนี้โดยอาศัยประสบการณตรงของการเปนผูซื้อมาพัฒนา

ระบบ โดยมองตามความนาจะเปน ในเรื่องของงานขาย การจัดทํ ารายงานตลอดจนการควบคุมสินคาคงคลัง และจะ
เนนการขายสง  โดยไมมีการขายหนาราน นั่นคือรับค ําสั่งซื้อทางโทรศัพท จดหมาย หรือโทรสาร เพื่อใหเขาใจระบบ
ไดอยางงาย ๆ

6.1.1 บุคคลที่เกี่ยวของกับงาน
ลํ าดับ ผูเกี่ยวของ ความสัมพันธตอขอมูล

1 ลูกคา สัง่ซื้ออยูที่บาน หรือที่ท ํางาน
2 ลกูจาง หรือพนักงานขาย ปอนขอมูลเขาเครื่องเมื่อมีการสั่งซื้อ หรือฝายอื่น ๆ
3 เจาของกิจการ รบัรายงานการขาย ตาง ๆ
4 ผูผลิตสินคา สงสินคาเขาราน
5 บริษัทรับสงของ เพราะบรษิัทนี้ไมสงของเอง แตจางบริษัทรับสงของมาชวยรับภาระ

 6.1.2 ระบบงานที่เปนอยู และสิ่งที่ผูใชตองการ
ผูเขยีนตองการใหระบบนี ้เปนตัวอยางการพัฒนาระบบที่ทํ าใหผูอานเขาใจไดงาย ๆ จึงควบคุมขอบเขตของ

งานใหอยูในเรื่องการขายสง โดยซื้อสินคาจากผูผลิต รอรับใบสั่งซื้อ  แลวสงสินคาใหลูกคาโดย จางบริษัทขนสงมารับ
ผดิชอบ และระบบนี้จะไมมองถึงหนี้สิน  เพราะความคิดที่วาลูกคาทุกคนเปนคนดี เมื่อรับของแลวจะสงเงินเขาบัญชี
ของบรษิัททันท ีและไมมีปญหาการสงคืนสินคา หรือเช็คเดงแนนอน

ผูเขยีนไดจ ําลองงานตาง ๆ ที่ระบบนี้ควรม ีเปนขอ ๆ ดวยภาษาของผูใชที่ไมมีความรูในคอมพิวเตอรและ
การเขยีนโปรแกรมมากเทาใด จึงไดรายการของความตองการดังตอไปนี้  13

13  ผูเขียนเคยอานหนังสือ 3 เลม คอื การประยุกตคอมพิวเตอรดานการเงิน และการบัญช ีประกอบดวย
ระบบบญัชสีินคาคงคลัง ระบบบัญชีเงินเดือน และระบบบัญชีแยกประเภท ของบริษัท คอมพิวเทคไมโครซิสเท็ม
จํ ากัด พ.ศ.2538 ซึง่มหีลายสวนที่มีอิทธิพลตอการออกแบบระบบกรณีศึกษาในบทนี้

ความตองการของผูใช
1. ขาย ดวยตัวแปรชุดเดียว
2. ขาย ดวยตัวแปรหลายชุด
3. ขาย ดวยตัวแปรอาเรย
4. พมิพรายงานการขาย ตามวันที่
5. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
6. พมิพขอมูลใบสั่งสินคา ตามรหัสพนักงานขาย
7. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
8. พมิพสรุปยอดขายสินคาแตละชนิด ตามเดือน
9. พมิพสรุปยอดขายสินคาแตละประเภท ตามเดือน
10. ปรบัปรุงแฟมตาง ๆ อยางงาย ๆ
11. พมิพขอมูลจากแฟมอยางงาย ๆ
12. ยกเลิกใบสั่งสินคา
13. ยกเลกิสนิคาในใบสั่งสินคาบางรายการ
14. ปรบัปรุงใบสั่งสินคาใหถูกตอง
15. พมิพขอมูลการขายสินคาในรอบเดือน ตามประเภทสินคา
16. พมิพสรุปคาขนสงใหบริษัทขอสง ตามบริษัทขนสง
17. พมิพสรุปยอดขายสินคา ตามผูผลิตสินคา
18. พมิพสรุปยอดขายสินคา ตามผูผลิตสินคา และเดือน
19. สัง่เพิม่เงนิเดือนพนักงานทุกคน ตามเปอรเซนตที่ตองการ
20. พมิพสถติกิารใชบริการบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
21. พมิพรายการสินคาที่ขายด ีจากมากไปนอย ตามชวงวันที่
22. พมิพใบสั่งสินคาที ่สงไมทันก ําหนด ตามเดือน
23. พมิพจดหมายขอโทษ ที่สงสินคาไมทันตามกํ าหนด ตามเลขที่ใบสั่งสินคาที่ตองการ
24. พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
25. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
26. เพิม่ยอดสินคาเขามาในคลังสินคา
27. รบัสนิคารายการใหมเขามาในคลังสินคา
28. พมิพรายการสินคาที่กํ าลังสั่งซื้อ เพิ่มปจจุบัน
29. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้
30. พมิพรายการสินคาที่ถึงจุดสั่งซื้อ ตามผูผลิตสินคา และยังไมไดสั่งซื้อเพิ่ม

&  6.2 โครงสรางขอมูล
โครงสรางขอมูลตอไปนี ้ถูกออกแบบใหรองรับงานตามความตองการขางตน  โดยแยกแฟมออกตาม

ลกัษณะเฉพาะของขอมูล และเชื่อมแฟมเขาดวยกันดวยฟลดหลักหลาย ๆ ฟลดจนเปนฐานขอมูลที่เก็บขอมูลไดไมซํ ้า
ซอน มคีวามกระชับและเรียกขอมูลมาใชรวมกันได
แฟมทีเ่กี่ยวของกับระบบงานขายและสินคาคงคลัง
1. แฟมสั่งสินคา    ORDER.DBF
    แฟมดรรชน ี ORDER1.NTX    จดัเรยีงตาม ORDER,CUST,EMPL

1.1 รหัสใบสั่ง       *ORDER         9(5)
  1.2 รหัสลูกคา       *CUST          9(4)
  1.3 รหัสพนักงาน      *EMPL          9(4)
  1.4 สถานที่จัดสง     SHIPADDR      X(40)
  1.5 วันที่สั่งสินคา     DATEORD       X(8)
  1.6 วนัที่ตองการสินคา  DATEREQ       X(8)
  1.7 วนัที่จัดสงสินคา   DATESHIP      X(8)
  1.8 บริษัทจัดสง      SHIPPER       9(2)
  1.9 คาขนสง          FREIGHT       9(5)V99

  ตวัอยางขอมูล
  95001 1009 3804 444 พระบาท อ.เมือง ล ําปาง 52000  12/09/95 25/09/95 81 18/09/95
  95002 1012 3844 43 บานผา อ.เกาะคา ล ําปาง 52130   19/09/95 26/09/95 83 30/09/95
  95003 1021 3801 21 บานทอง อ.เมือง เลย 44000       21/09/95 28/09/95 81 09/10/95
  95004 1033 3816 5 ต.คลองด ํา อ.เมือง ตาก 63000      25/09/95 06/10/95 82 15/10/95

2. แฟมสั่งสินคาละเอียด ORDERD.DBF
   แฟมดรรชน ี ORDERD1.NTX   จดัเรยีงตาม ORDER,PRO
  ORDERD2.NTX   จดัเรยีงตาม PRO,ORDER
  2.1 รหัสใบสั่ง       *ORDER         9(5)
  2.2 รหัสสินคา       *PRO           9(4)
  2.3 ราคาขาย           OPRICE        9(5)V99
  2.4 ปริมาณขาย         OQUANT        9(4)
  2.5 สวนลด            ODISC         9(4)V99

  ตวัอยางขอมูล
  95001 1105    12.00   20   40.00
  95001 1116  200.50  100   50.00
  95002 1111    81.00   30   30.00
  95002 1121  950.00   10    0.00

3.  แฟมสินคา      PRO.DBF
    แฟมดรรชน ี PRO1.NTX      จดัเรียงตาม PRO
  3.1 รหัสสินคา       *PRO           9(4)
  3.2 รหัสผูผลิด       SUP           9(3)
  3.3 รหัสประเภท        CAT           9(2)
  3.4 ชื่อสินคา        PRONAME       X(40)
  3.5 หนวยที่ใช       PROUNIT       X(10)
  3.6 ราคาทุน           PROPRICE      9(5)V99
  3.7 จ ํานวนเหลือ        PROREST       9(4)
  3.8 จ ํานวนที่สั่งซื้อเพิ่ม  PROONORD 9(4)
  3.9 จดุสั่งซื้อ       REORDER       9(3)

  ตวัอยางขอมูล
  1101 511 14 ปากกา  โหล  25.00   50    0   10
  1102 509 13 กระดาษ รีม  75.00  141  300   150

4. แฟมพนักงาน    EMPL.DBF
    แฟมดรรชน ี EMPL1.NTX     จัดเรียงตาม EMPL
  4.1 รหัสพนักงาน      *EMPL          9(4)
  4.2 ชือ่พนักงาน       EMPLNAME      X(40)
  4.3 ทีอ่ยูพนักงาน    EMPLADDR      X(40)
  4.4 เงินเดือน         EMPLSAL       9(6)V99
  4.5 วนัเกิด           EMPLBORN      X(8)

  ตวัอยางขอมูล
  3801  นาย สมชาย ดีดี  44 ต.สบสงค อ.ดานเกวียน เลย 48000   6500.00  13/06/69
  3802  น.ส.งาน การดี   210 ต.คงคง อ.รักคํ า ล ําพูน 51000       8900.00  21/01/54
  3803  นาย ดง ไผดึก   158 ต.ฐานะ อ.มะระ ล ําปาง 52000       10000.00 09/12/65

5. แฟมลูกคา        CUST.DBF
    แฟมดรรชน ี CUST1.NTX     จดัเรียงตาม CUST
  5.1 รหัสลูกคา      *CUST          9(4)
  5.2 ชือ่ลูกคา        CUSTNAME      X(40)
  5.3 ที่อยูลูกคา     CUSTADDR      X(40)

  ตวัอยางขอมูล
  1001  นาย ไขคํ า ดงดึก    488 ต.ปาดง อ.บานสัน สมุทรสงคราม 16000
  1002  นาย คํ าด ีมีไชย     12 ต.คงคา อ.บานสัน สมุทรสงคราม 16000
  1003  น.ส.เดก็ขาว กากา   1000 ต.ฐานะ อ.มะระ ล ําปาง 52000

6.  แฟมบริษัทรับสงของ   SHIPPER.DBF
     แฟมดรรชน ี SHIPPER1.NTX  จดัเรียงตาม SHIPPER
  6.1 รหัสบริษัทรับสง  *SHIPPER      9(2)
  6.2 ชื่อบริษัทรับสง  SHIPNAME      X(40)

  ตวัอยางขอมูล
  81  แสงตะวันขนสง
  82  แสงหลาขนสง
  83  ค ําดีรักบริการขนสง

7. แฟมประเภทสินคา   CAT.DBF
    แฟมดรรชน ี CAT1.NTX      จดัเรียงตาม CAT
  7.1 รหัสประเภท       *CAT          9(2)
  7.2 ชื่อประเภทสินคา  CATNAME       X(40)

  ตวัอยางขอมูล
  11 เครื่องดื่ม
  12 เครื่องเขียน
  13 ผักและผลไม

8.  แฟมผูผลิต      SUP.DBF
    แฟมดรรชน ี SUP1.NTX      จดัเรียงตาม SUP ตวัอยางขอมูล
  8.1 รหัสผูผลิต      *SUP           9(3) 501   กาแฟเขาชอง จํ ากัด
  8.2 ชื่อผูผลิต       SUPNAME       X(40) 502   ซพีี ประเทศไทย

&   6.3 วิเคราะหระบบงานเบื้องตน
การวเิคราะหระบบงานครั้งนี ้ เนนการจัดทํ าโปรแกรม จึงศึกษาความความตองการ แลวน ําความตองการ

มาจดักลุมเปนระบบเล็ก ๆ   จากนั้นนํ ามาเปนแนวในการจัดทํ าเมนู  เพื่อแบงกลุมของโปรแกรม  ท ําใหมองภาพไดชัด
เจนโดยมคีวามสัมพันธโดยตรงมาจาก สิ่งที่ตองการ

การวเิคราะหระบบงานแบบนี้ไมสามารถน ําไปใชงานจริงได เพราะยังขาดสิ่งสํ าคัญตาง ๆ อีกมาก เชนการ
ศกึษาความเปนไปไดของระบบ  การจัดทํ าผังงาน การศึกษาทิศทางการใชและการเปลี่ยนแปลงของขอมูล เปนตน

6.3.1 จัดระบบยอยจากความตองการของผูใช
ระบบขาย
   1. ขาย ดวยตัวแปรชุดเดียว
   2. ขาย ดวยตัวแปรหลายชุด
   3. ขาย ดวยตัวแปรอาเรย

ระบบปรับปรุง
   1. ปรบัปรุงใบสั่งสินคาใหถูกตอง
   2. ยกเลิกใบสั่งสินคา
   3. ยกเลกิสินคาในใบสั่งสินคาบางรายการ
   4. สัง่เพิม่เงนิเดือนพนักงานทุกคน ตามเปอรเซนตที่ตองการ
   5. ปรบัปรุงแฟมตาง ๆ อยางงาย ๆ
ระบบรายงานยอดขาย
   1. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
   2. พมิพสรุปยอดขายสินคาแตละชนิด ตามเดือน
   3. พมิพสรุปยอดขายสินคาแตละประเภท ตามเดือน
   4. พิมพสรุปยอดขายสินคา ตามผูผลิตสินคา
   5. พมิพสรุปยอดขายสินคา ตามผูผลิดสินคา และเดือน
   6. พมิพรายงานการขาย ตามวันที่
   7. พมิพขอมูลใบสั่งสินคา ตามรหัสพนักงานขาย
   8. พมิพขอมูลการขายสินคาในรอบเดือน ตามประเภทสินคา
ระบบควบคุมสินคาคงคลัง
   1. เพิม่ยอดสินคาเขามาในคลังสินคา
   2. รบัสนิคารายการใหมเขามาในคลังสินคา
   3. พมิพรายการสินคาที่ ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
   4. พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
   5. พมิพรายการสินคาที่ก ําลังสั่งซื้อ เพิ่มปจจุบัน
   6. พมิพรายการสินคาที่ถึงจุดสั่งซื้อ ตามผูผลิตสินคา และยังไมไดสั่งซื้อเพิ่ม
   7. พมิพรายการสินคาที่ขายด ีจากมากไปนอย ตามชวงวันที่
ระบบรายงานอื่น
   1. พมิพขอมูลจากแฟมอยางงาย ๆ
   2. พมิพสรุปคาขนสงใหบริษัทขอสง ตามบริษัทขนสง
   3. พมิพสถติกิารใชบริการบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
   4. พมิพใบสั่งสินคาที ่สงไมทันก ําหนด ตามเดือน
   5. พมิพจดหมายขอโทษ ที่สงสินคาไมทันตามกํ าหนด ตามเลขที่ใบสั่งสินคาที่ตองการ
   6. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
   7. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้

6.3.2 ออกแบบเมนูตามระบบยอย
เมือ่ไดมองเหน็กลุมงานซึ่งถูกแยกเปนระบบยอยแลว   เพื่อใหการออกแบบเมนูงายขึ้น  ในกรณีนี้จึงขอออก

แบบตวัเลอืกตามระบบยอย  แตควรแตงคํ าพูดใหกระชับขึ้น เพื่อจะไดไมยาวจนเกินไปในการทํ าเมนู
เมนูหลัก
1. ระบบขาย
2. ระบบปรับปรุง
3. ระบบรายงานยอดขาย
4. ระบบควบคุมสินคาคงคลัง
5. ระบบรายงานอื่น
6. เลกิการท ํางาน
1. เมนูระบบขาย
   1. ขาย ดวยตัวแปรชุดเดียว
   2. ขาย ดวยตัวแปรหลายชุด
   3. ขาย ดวยตัวแปรอาเรย
   4. กลับเมนูหลัก
2. เมนูระบบปรับปรุง
   1. ปรับปรุงใบสั่งสินคา
   2. ยกเลิกใบสั่งสินคาทั้งใบ
   3. ยกเลิกสินคาในใบสั่งสินคา
   4. เพิม่เงินเดือนพนักงาน
   5. ปรับปรุงแฟมตาง ๆ
   6. กลับเมนูหลัก
3. ระบบรายงานยอดขาย
   1. พมิพสรุปยอดขายของสินคา ตามเดือน
   2. พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา
   3. พิมพสรุปยอดขายของสินคา ตามผูผลิตสินคา และเดือน
   4. พมิพสรุปยอดขายของสินคาแตละประเภท ตามเดือน
   5. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
   6. พมิพสรุปยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย
   7. พมิพสรุปยอดขายสินคา ตามประเภทสินคา
   8. พมิพรายการขายสินคาแตละรายการ ตามวันที่
   9. กลับเมนูหลัก

4. ระบบควบคุมสินคาคงคลัง
   1. ซือ้สนิคาเพิ่ม เมื่อสินคาถึงจุดสั่งซื้อ
   2. ซือ้สนิคารายการใหม มาจํ าหนาย
   3. พมิพรายการสินคาที่ ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
   4. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ ตามผูผลิต และยังไมไดสั่งเพิ่ม
   5. พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
   6. พมิพรายการสินคาที ่กํ าลังสั่งซื้อเพิ่มปจจุบัน
   7. พมิพรายการสินคาที ่ขายดี จากมากไปนอย ตามชวงวันที่
   8. กลับเมนูหลัก
5. ระบบรายงานอื่น
   1. พมิพสรุปคาขนสงใหบริษัทขอสง ตามเดือน
   2. พมิพสถติกิารใชบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
   3. พมิพจดหมายขอโทษ ที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสั่งสินคา
   4. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
   5. พมิพเลขที่ใบสั่งสินคา ที่สงไมทันก ําหนด ตามเดือน
   6. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้
   7. พมิพขอมูลจากแฟมอยางงาย ๆ
   8. กลับเมนูหลัก

6.3.3 หลักการตั้งชื่อโปรแกรม
ส ําหรบัระบบงานขาย และสินคาคงคลัง ไดตั้งชื่อโปรแกรมตามลํ าดับในเมนู  เพื่อสื่อใหเขาใจไดงายโดย

มองล ําดับของตัวเลือกตามเมนู  ส ําหรับชื่อโปรแกรมก ําหนดไว 6 หลัก
หลกัที ่1 ใหเปนตัว S ซึ่งยอมาจาก SALE
หลักที ่2 และ 3 เปนล ําดับในเมนูหลัก
หลกัที ่4 และ 5 เปนล ําดับในเมนูยอย

โปรแกรมตาง ๆ ในระบบทะเบียนมีดังนี้
MENU.PRG : เมนหูลกัส ําหรับระบบงานขาย และสินคาคงคลัง ประกอบดวย 10 โปรแกรมยอย
MENUALL : โปรแกรมหลักที่เรียกเมนูตาง ๆ มาใหเลือก
PRESSDOWN : กระท ําเมื่อกดปุมลูกศรลง เพื่อปรับปรุงคาสูงสุด และตํ ่าสุดของตัวเลือก
PRESSUP : กระท ําเมื่อกดปุมลูกศรขึ้น เพื่อปรับปรุงคาสูงสุด และตํ ่าสุดของตัวเลือก
WRITEMENU : เขยีนเมนูใหมทุกครั้งที่กดลูกศร
ALERTNEW : ใชแทนค ําสั่ง ALERT ท ําใหไดตัวเลือกมากกวา 4 และเปลี่ยนสีได
ASKPRT : สอบถามการปลายทางของผลลัพธ
ASKUPD : ถามยํ ้าการปรับปรุงขอมูล

HELPFORM : แสดงขอความชวยเหลืออยางงาย ๆ เมื่อกดปุมฟงกชัน F1
HELPREADVAR : แสดงคาที่เปนไปไดของฟลดที่ก ําลังรอรับคา
QUITPROC : สัง่เลกิท ํางานของโปรแกรมทั้งหมด
SALE1.PRG : ระบบขาย ประกอบดวย 3 โปรแกรมยอย
S0101 : ขาย ดวยตัวแปรชุดเดียว
S0102 : ขาย ดวยตัวแปรหลายชุด
S0103 : ขาย ดวยตัวแปรอาเรย
SALE2.PRG : ระบบปรับปรุง ประกอบดวย 5 โปรแกรมยอย
S0201 : ปรับปรุงใบสั่งสินคา
S0202 : ยกเลิกใบสั่งสินคาทั้งใบ
S0203 : ยกเลิกสินคาในใบสั่งสินคา
S0204 : เพิม่เงินเดือนพนักงาน
S0205 : ปรับปรุงแฟมตาง ๆ
SALE3.PRG : ระบบรายงานยอดขาย ประกอบดวย 8 โปรแกรมยอย
S0301 : พมิพสรุปยอดขายของสินคา ตามเดือน
S0302 : พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา
S0303 : พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา และเดือน
S0304 : พมิพสรุปยอดขายของสินคาแตละประเภท ตามเดือน
S0305 : พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
S0306 : พมิพสรปุยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย
S0307 : พมิพสรุปยอดขายสินคา ตามประเภทสินคา
S0308 : พมิพรายการขายสินคาแตละรายการ ตามวันที่
SALE4.PRG ระบบควบคุมสินคาคงคลัง ประกอบดวย 7 โปรแกรมยอย
S0301 : ซือ้สนิคาเพิ่ม เมื่อสินคาถึงจุดสั่งซื้อ
S0302 : ซือ้สนิคารายการใหม มาจํ าหนาย
S0303 : พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
S0304 : พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ ตามผูผลิต และยังไมไดสั่งเพิ่ม
S0305 : พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
S0306 : พมิพรายการสินคาที่ กํ าลังสั่งซื้อเพิ่มปจจุบัน
S0307 : พมิพรายการสินคาที ่ขายดี จากมากไปนอย ตามชวงวันที่

SALE5.PRG ระบบรายงานอื่น ประกอบดวย 7 โปรแกรมยอย
S0301 : พมิพสรุปคาขนสงใหบริษัทขอสง ตามเดือน
S0302 : พมิพสถติกิารใชบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
S0303 : พมิพจดหมายขอโทษ ที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสั่งสินคา
S0304 : พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
S0305 : พมิพเลขที่ใบสั่งสินคา ที่สงไมทันก ําหนด ตามเดือน
S0306 : พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้
S0307 : พมิพขอมูลจากแฟมอยางงาย ๆ

6.3.4 วิธีการแปลโปรแกรม
การแปลโปรแกรมส ําหรับกรณีศึกษาครั้งนี้  ใชคํ าสั่ง RMAKE ในการแปลโปรแกรม โดยระบุชื่อโปรแกรม

หลัก และโปรแกรมยอยทั้งหมด  ไวในแฟม MENU.RMK
การสรางแฟม MENU.RMK
//  MENU.RMK
//  ========
.PRG.OBJ:

          CLIPPER $< /M /N
.OBJ.EXE:
       SET RTLINKCMD=/POSI
       RTLINK $**, $@ ;
MENU.OBJ:     MENU.PRG
SALE1.OBJ:    SALE1.PRG
SALE2.OBJ:    SALE2.PRG
SALE3.OBJ:    SALE3.PRG
SALE4.OBJ:    SALE4.PRG
SALE5.OBJ:    SALE5.PRG
MENU.EXE: MENU.OBJ  SALE1.OBJ SALE2.OBJ SALE3.OBJ SALE4.OBJ SALE5.OBJ

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


294                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

&  6.4 โปรแกรมในระบบงานขาย และสินคาคงคลัง
โปรแกรมของระบบนี ้ ไดจัดทํ าเปนโปรแกรมยอยไวภายนอก โดยมีโปรแกรม MENU.PRG คอยเรียก

โปรแกรมเล็ก ๆ ที่อยูในโปรแกรมยอยอีกครั้งหนึ่ง  ส ําหรับโปรแกรมยอยภายนอกนั้น แตละโปรแกรมยอยคือระบบ
ยอยระบบหนึง่ เพราะทํ าใหงายตอการปรับปรุงแกไขและคนหา หากประมวลผลตามตัวเลือกแลวผิดพลาด

โปรแกรมยอยจากภายนอกม ี 5 โปรแกรมคือ SALE1.PRG จนถึง SALE5.PRG  รวมโปรแกรมทั้งหมดใน
ระบบนี้เปน 6  โปรแกรม  โดยใหโปรแกรมยอยทั้ง 5 นี้ควบคุมระบบยอย 5 ระบบ

1. ระบบขาย
2. ระบบปรับปรุง
3. ระบบรายงาน
4. ระบบคงคลัง
5. ระบบอื่น

6.4.1 โปรแกรมตาง ๆ ใน MENU.PRG
โปรแกรมใน MENU.PRG นี้มีโปรแกรมยอยทั้งหมด 10 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับเมนู และ

โปรแกรมสนับสนุน เชน โปรแกรมสอบถามปลายทางของผลลัพธ  โปรแกรมถามยํ้ าการปรับปรุงขอมูล  โปรแกรม
แสดงสวนชวยเหลือแบบตาง ๆ   โปรแกรมสั่งเลิกการท ํางาน เปนตน

โปรแกรมเมนูในโปรแกรม MENU.PRG นี ้ ใชหลักการในการตรวจสอบการกดลูกศร ท ําใหเมนูแบบนี้
สามารถแสดงนาฬิกาที่เดิน ขณะรอเลือกเมนูได   เมนูแบบนี้ใหประสิทธิภาพในการใชการ สามารถเพิ่มการทํ างาน
ตาง ๆ  ไดอีกมาก  โดยโปรแกรมเมนูนี้เรียกโปรแกรมยอยจากภายในอีก 3 โปรแกรมคือ PRESSDOWN ,  PRESSUP
และ WRITEMENU   การนํ าโปรแกรมนี้ไปปรับปรุงแกไขและน ําไปใชทํ าไดโดยงาย  เพราะโปรแกรมนี้จะตรวจสอบ
การกดปุมลกูศร แลวท ําการจอภาพใหมทุกครั้ง  แตถากดปุม ENTER จะประมวลผลตามเงื่อนไขที่ระบุตามตัวเลือก
นั้น

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม MAIN.PRG ทั้งหมด 10 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.1 ถงึ ตวัอยางที่ 6.8  (บางโปรแกรมสั้นและมีการท ํางานคลายกัน จึงนํ ามารวมในตัวอยางเดียวกัน)
: ตัวอยางที่ 6.1
&   ผลลัพธ
// โปรแกรมหลัก ของระบบขาย และสินคาคงคลัง เพื่อจัดทํ าเมนูหลักและเมนูยอยใน
// โปรแกรมเดียวกัน เมนูมีลักษณะเปนแบบดึงลง และมีนาฬิกาวิ่งตลอดเวลา
// ซึง่เมนแูบบนี้สามารถปรับปรุง แกไขใหมีประสิทธิภาพไดอีกมากมาย
// และมกีารเรียกโปรแกรมยอยเพื่อจัดท ําเมนูอีก 3 โปรแกรมคือ
// 1. ท ําหนาที่เปลี่ยนคาเมื่อกดลงดวย โปรแกรมยอย PRESSDOWN
// 2. ท ําหนาที่เปลี่ยนคาเมื่อกดขึ้นดวย โปรแกรมยอย PRESSUP
// 3. ท ําหนาที่เขียนเมนูใหมเมื่อเลื่อนตัวเลือกดวย โปรแกรมยอย WRITEMENU

PROCEDURE MENUALL
   SET KEY 28 TO HELPFORM  // ควบคมุปุม F1 เพื่อแสดงขอมูลเกี่ยวกับผูเขียนโปรแกรม
   SET KEY -1 TO HELPREADVAR // ควบคมุปุม F2 เพื่อเลือกขอมูลไดอยางงาย ๆ
   SET KEY -9 TO QUITPROC    // ควบคมุปุม F10 เพื่อถามการเลิกทํ างาน
   SET WRAP ON
   SET DELIMITERS TO '[ ]'
   SET DELIMITERS ON
   SET EPOCH TO 1960
   SET CURSOR OFF
   ARMAIN = 6
   ARCM = ARRAY(ARMAIN)
   ARCM[1] = [1.ระบบขาย]
   ARCM[2] = [2.ระบบปรับปรุง]
   ARCM[3] = [3.ระบบรายงาน]
   ARCM[4] = [4.ระบบคงคลัง]
   ARCM[5] = [5.ระบบอื่น]
   ARCM[6] = [6.เลกิงาน]
   ARS1 = 4
   ARC1 = ARRAY(ARS1)
   ARC1[1] = [1. ขาย ดวยตัวแปรชุดเดียว]
   ARC1[2] = [2. ขาย ดวยตัวแปรหลายชุด]
   ARC1[3] = [3. ขาย ดวยตัวแปรอาเรย]
   ARC1[4] = [4. กลับเมนูหลัก]
   ARS2 = 6
   ARC2 = ARRAY(ARS2)
   ARC2[1] = [1. ปรับปรุงใบสั่งสินคา]
   ARC2[2] = [2. ยกเลกิใบสั่งสินคาทั้งใบ]
   ARC2[3] = [3. ยกเลกิสินคาในใบสั่งสินคา]
   ARC2[4] = [4. เพิม่เงินเดือนพนักงาน]
   ARC2[5] = [5. ปรบัปรุงแฟมตาง ๆ]
   ARC2[6] = [6. กลับเมนูหลัก]
   ARS3 = 9
   ARC3 = ARRAY(ARS3)

   ARC3[1] = [1. พมิพสรุปยอดขายของสินคา ตามเดือน]
   ARC3[2] = [2. พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา]
   ARC3[3] = [3. พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา และเดือน]
   ARC3[4] = [4. พมิพสรุปยอดขายของสินคาแตละประเภท ตามเดือน]
   ARC3[5] = [5. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่]
   ARC3[6] = [6. พมิพสรปุยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย]
   ARC3[7] = [7. พมิพสรุปยอดขายสินคา ตามประเภทสินคา]
   ARC3[8] = [8. พมิพรายการขายสินคาแตละรายการ ตามวันที่]
   ARC3[9] = [9. กลับเมนูหลัก]
   ARS4 = 8
   ARC4 = ARRAY(ARS4)
   ARC4[1] = [1. ซือ้สนิคาเพิ่ม เมื่อสินคาถึงจุดสั่งซื้อ]
   ARC4[2] = [2. ซือ้สนิคารายการใหม มาจํ าหนาย]
   ARC4[3] = [3. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา]
   ARC4[4] = [4. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ ตามผูผลิต และยังไมไดสั่งเพิ่ม]
   ARC4[5] = [5. พมิพรายการสินคาที ่เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ]
   ARC4[6] = [6. พมิพรายการสินคาที่ กํ าลังสั่งซื้อเพิ่มปจจุบัน]
   ARC4[7] = [7. พมิพรายการสินคาที ่ขายดี จากมากไปนอย ตามชวงวันที่]
   ARC4[8] = [8. กลับเมนูหลัก]
   ARS5 = 8
   ARC5 = ARRAY(ARS5)
   ARC5[1] = [1. พมิพสรุปคาขนสงใหบริษัทขนสง ตามเดือน]
   ARC5[2] = [2. พมิพสถติกิารใชบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่]
   ARC5[3] = [3. พมิพจดหมายขอโทษ ที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสั่งสินคา]
   ARC5[4] = [4. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่]
   ARC5[5] = [5. พมิพเลขที่ใบสั่งสินคา ที่สงไมทันก ําหนด ตามเดือน]
   ARC5[6] = [6. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้]
   ARC5[7] = [7. พมิพขอมูลจากแฟมอยางงาย]
   ARC5[8] = [8. กลับเมนูหลัก]
   OPT = 1                                 // ก ําหนดตัวเลือกเริ่มตน
   TOPLMT = OPT                            // ก ําหนดบรรทัดบนสุดของเมนูปจจุบัน
   BOTTOMLMT = ARMAIN                      // ก ําหนดบรรทัดลางสุดของเมนูปจจุบัน
   ALLSUB = ARMAIN                         // จ ํานวนเมนูยอยทั้งหมด

   DO WRITEMENU                            // พิมพเมนูครั้งแรก
   DO WHILE LASTKEY() != 27
      @ 1,5 SAY TIME()                     // แสดงเวลาตลอดเวลา
      INKEY(0.2)                           // หยุดรับแปนพิมพทุก 0.2 วินาที
      IF LASTKEY() != 0
         // กรณกีดปุมใด ๆ จะมาตรวจสอบในสวนนี้
         DO CASE
            CASE LASTKEY() = 27     // กด ESC
                 IF OPT <= ALLSUB
                    DO QUITPROC
                 ELSE
                    OPT = (OPT - MOD(OPT,10))/10
                 ENDIF
            CASE LASTKEY() = 4      // กดขวา
                 IF OPT <= ALLSUB           // ตวัเลือกในเมนูหลัก
                    DO PRESSDOWN
                 ELSE
                    IF (OPT-MOD(OPT,10))/10 = ALLSUB // ขามไปเมนูทางขวาหรือเริ่มใหม
                       OPT = 11
                    ELSE
                       OPT = (OPT-MOD(OPT,10))+11
                    ENDIF
                 ENDIF
            CASE LASTKEY() = 19     // กดซาย
                 IF OPT < 10                    // ตวัเลือกในเมนูหลัก
                    DO PRESSUP
                 ELSE
                    IF (OPT-MOD(OPT,10))/10 = 1 // ขามไปเมนูทางซายหรือสุดทาย
                       OPT = ALLSUB*10+1
                    ELSE
                       OPT = (OPT-MOD(OPT,10))-9
                    ENDIF
                 ENDIF

           CASE LASTKEY() = 24     // กดลง
                DO PRESSDOWN
           CASE LASTKEY() = 5      // กดขึ้น
                DO PRESSUP
           // กรณกีด ENTER จะมาเลือกประมวลผลในสวนนี้
           CASE LASTKEY() = 13     // กดเอ็นเทอร (ENTER)
                DO CASE
                   CASE OPT >= 1 .AND. OPT <= ARMAIN -1
                        OPT = OPT * 10 + 1
                   CASE OPT = ARMAIN
                        @ 20,5 SAY "BYE BYE BYE"
                        DO QUITPROC
                   CASE OPT=10+ARS1 .OR. OPT=20+ARS2 .OR. ;
                        OPT=30+ARS3 .OR. OPT=40+ARS4 .OR. ;
                        OPT=50+ARS5                   // กลับเมนูหลัก
                        OPT = (OPT - MOD(OPT,10))/10
                   OTHERWISE  // กระท ําตัวเลือก
                        ATTAGX = LTRIM(STR(MOD(OPT,10),1))
                        ATSUBX = LTRIM(STR((OPT - MOD(OPT,10))/10,1))
                        SET CURSOR ON
                        CLS
                        DO &("S0" + ATSUBX + "0" + ATTAGX)
                        SET CURSOR OFF
               ENDCASE
           ENDCASE
           // ตัง้คา TOPLMT และ BOTTOMLMT ใหม
           DO CASE
              CASE OPT <10
                 TOPLMT = 1       ; BOTTOMLMT = ARMAIN
              CASE OPT >= 11 .AND. OPT<=10+ARS1
                 TOPLMT = 11      ; BOTTOMLMT = 10+ARS1
              CASE OPT >= 21 .AND. OPT<=20+ARS2
                 TOPLMT = 21      ; BOTTOMLMT = 20+ARS2

              CASE OPT >= 31 .AND. OPT<=30+ARS3
                 TOPLMT = 31      ; BOTTOMLMT = 30+ARS3
              CASE OPT >= 41 .AND. OPT<=40+ARS4
                 TOPLMT = 41      ; BOTTOMLMT = 40+ARS4
              CASE OPT >= 51 .AND. OPT<=50+ARS5
                 TOPLMT = 51      ; BOTTOMLMT = 50+ARS5
              CASE OPT = 61
                 TOPLMT = 1; BOTTOMLMT = ARMAIN; OPT = 6
           ENDCASE
           DO WRITEMENU           // เขียนเมนูใหม
           KEYBOARD CHR(0)
           INKEY()                // เพือ่ตัง้คาของ LASTKEY() ใหม
       ENDIF
   ENDDO
RETURN
: ตัวอยางที่ 6.2
&   ผลลัพธ
// ใชประกอบโปรแกรมเมนูหลัก โดยกระท ําเมื่อกดปุมลง หรือขึ้น ในการเลือกตัวเลือก
// โดยมเีปาหมายคือปรับปรุงคาของตัวเลือกปจจุบันใหเพื่อขึ้นหรือลดลง
// กระท ําเมื่อกดปุมลงในเมนู
PROCEDURE PRESSDOWN
   IF OPT = BOTTOMLMT // ถาถงึตัวลางสุดแลวยังกดลง
      OPT = TOPLMT   // ให OPT ไปที่ตัวแรก คือ TOPLMT (Top limit)
   ELSE
      OPT = OPT + 1
   ENDIF
RETURN
// กระท ําเมื่อกดปุมขึ้นในเมนู
PROCEDURE PRESSUP
   IF OPT = TOPLMT    // ถาถึงตัวแรกแลวยังกดขึ้น
      OPT = BOTTOMLMT // ให OPT ไปยังตัวสุดทาย
   ELSE
      OPT = OPT - 1
   ENDIF
RETURN

: ตัวอยางที่ 6.3
&   ผลลัพธ
// ใชประกอบโปรแกรมหลัก เพื่อท ําการเขียนจอเมนูใหถูกตองโดยใชคาตัวเลือกปจจุบัน
// การเขยีนเมนูของโปรแกรมนี ้จะเขียนใหมทุกครั้งที่มีการกดปุมใด ๆ
// โปรแกรมยอยนี้จึงใชสํ าหรับเขียนหนาเมนูใหมเมื่อมีการกดแปนพิมพ
PROCEDURE WRITEMENU
   SET COLOR TO W/B
   CLS ; SETPOS(2,0)
   XCOL = ARRAY(ARMAIN)
   FOR J = 1 TO ARMAIN                // พิมพเมนูหลัก
      XCOL[J] = COL() + 1             // เกบ็หลักของเมนูยอยแตละเมนู
      @ 2,XCOL[J] SAY ARCM[J]
   NEXT
   ATTAG = MOD(OPT,10)          // หางของตัวเลือก
   ATSUB = (OPT - ATTAG)/10     // เมนูยอยที่
   IF ATSUB > 0
      I = 4                              // ระบุเลขที่บรรทัดเริ่มตน
      FOR J = I TO I - 1 + &("ARS"+LTRIM(STR(ATSUB,1)))
         @ J,XCOL[ATSUB] SAY &("ARC"+LTRIM(STR(ATSUB,1)))[J-I+1]
      NEXT
      TEMPAR = &("ARC"+LTRIM(STR(ATSUB,1)))
   ENDIF
  SET COLOR TO B/W
  IF OPT > 10                                   // สัง่เลือกตัวเลือกยอย
     @ I-1+ATTAG,XCOL[ATSUB]  SAY TEMPAR[ATTAG] // พมิพตัวเลือกปจจุบัน
     SET COLOR TO W/N
     @ 2,XCOL[ATSUB] SAY ARCM[ATSUB] // พมิพหัวของเมนูยอย
  ELSE
     @ 2,XCOL[ATTAG] SAY ARCM[ATTAG]     // พมิพหัวของเมนูยอยในขณะเลือกหัว
  ENDIF
  SET COLOR TO W/B
RETURN

: ตัวอยางที่ 6.4

&   ผลลัพธ
// ใชแทนฟงกชัน ALERT เพราะทํ าใหใชตัวเลือกไดมากกวา 4 ตัวเลือก และแสดง
// ตวัเลอืกในลักษณะ PULLDOWN CHOICE แลวสามารถระบุสีไดอยางอิสระ
// โดยไมท ําใหสถานะเดิมของสีเปลี่ยนไป เพราะมีการจัดเก็บไวแลว
FUNCTION ALERTNEW
PARAMETER HEADING,CHOICEAR
   CNTAR = 0
   AEVAL(CHOICEAR,{|ELE|CNTAR:=CNTAR+1})
   OLDCOLOR = SETCOLOR()
   SAVE SCREEN TO SCR
   CLS
   SET COLOR TO W/BR
   @ 10,20,12+CNTAR,60 BOX "+-+|+-+| "
   @ 11,21 SAY PADC(HEADING,38)
   FOR I = 1 TO CNTAR
       @ 11+I,30 PROMPT PADR(LTRIM(STR(I))+". "+CHOICEAR[I],30)
   NEXT
   MENU TO VALNEW
   SETCOLOR(OLDCOLOR)
   RESTORE SCREEN FROM SCR
RETURN (VALNEW)
: ตัวอยางที่ 6.5
&   ผลลัพธ
// ตวัอยางนี้ม ี2 ฟงกชันคือ ฟงกชันเพื่อเลือกปลายทางของผลลัพธ
// และฟงกชันถามยํ้ าการปรับปรุงขอมูล เมื่อมีการแกไขขอมูล
// ใหเสียงประกอบการเลือกทุกครั้ง
// ใชถามวาจะพิมพผลลัพธออกทางใด
FUNCTION ASKPRT()
   ASKPRT = ALERTNEW("ผลลพัธของรายงาน",{'จอภาพ','เครื่องพิมพ'})
   IF ASKPRT = 2
      TONE(500,5)
   ELSE
      TONE(800,8)
   ENDIF

RETURN (ASKPRT)
// ใชถามวาจะยอมรับการปรับปรุงขอมูลหรือไม
FUNCTION ASKUPD()
   ASKUPD = ALERTNEW("การปรบัปรุงขอมูล",{'ยอมรับการปรับปรุง','ยกเลิกการปรับปรุง'})
   IF ASKPRT = 2
      TONE(500,5)
   ELSE
      TONE(800,8)
   ENDIF
RETURN (ASKUPD)
: ตัวอยางที่ 6.6
&   ผลลัพธ
// พมิพขอมูลของผูพัฒนาโปรแกรมทางจอภาพ หากกดปุมฟงกชัน F1
// เพราะระบุใน MENUALL วา SET KEY 28 TO HELPFORM
PROCEDURE HELPFORM
   SAVE SCREEN TO SCR
   CLS
   ? "                วนัที่ :",DTOC(DATE())
   ? "               เวลา :",TIME()
   ? "====================================="
   ? "โปรแกรมนี้ถูกพัฒนาโดย. อ.บุรินทร รุจจนพันธุ"
   ? "เริ่มพัฒนาเมื่อ 8 กมุภาพันธ 2545"
   ? "โปรแกรมนี้เปนรุนที ่2.200202"
   ? "หากสนใจ หรือตองการตัวโปรแกรม ติดตอขอรับฟรีไดที"่
   ? "วทิยาลัยโยนก ล ําปาง 52000"
   ? "TEL.(054)265170 EXT.210   FAX.(054)265184"
   INKEY(5)
   RESTORE SCREEN FROM SCR
RETURN

: ตัวอยางที่ 6.7
&   ผลลัพธ
// ฟงกชนันีช้วยใหการใสคาในตัวแปรบางตัวแปรถูกตอง เพราะจะแสดง
// คาทีเ่หมาะสมกับฟลดนั้นมาใหเลือก โดยการอานคาจากแฟมมาแสดงใหดู
// แตจะตองระบุใหชัดเจนวาจะน ําฟลดใดมาใชบาง
// โปรแกรมนี้จึงแสดงขอมูลที่เปนไปไดส ําหรับตัวแปรที่กํ าลังรอรับคา
FUNCTION HELPREADVAR
  OLDALIAS = ALIAS() // เก็บพื้นที่ท ํางานเดิม
  SHOWAR = {}
  DO CASE
     CASE LEFT(READVAR(),4) = "_PRO" // เพราะอาจเปน _PRO[1] ได
          IF OLDALIAS != "PRO"
             USE PRO NEW // ถาเปดแลวตองไมเปดอีก
          ENDIF
          GO TOP
          WHILE !EOF()
              AADD(SHOWAR,STR(PRO)+" | "+PRONAME)
              SKIP
          END
          IND := ACHOICE(2,61,5,78,SHOWAR)
          IF IND > 0 // ปองกันการใชปุม ESC
             IF RIGHT(READVAR(),1) = "]"
                _PRO[J] = VAL(LEFT(SHOWAR[IND],4))
             ELSE
                _PRO = VAL(LEFT(SHOWAR[IND],4))
             ENDIF
          ENDIF
          IF OLDALIAS != "PRO" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
             CLOSE             // ในจุดที่ไมไดเปดแฟมนี้
          ENDIF
     CASE READVAR() = "_ORDER"
          SAVE SCREEN TO SCR
          IF OLDALIAS != "ORDER"
             USE ORDER NEW // ถาเปดแลวตองไมเปดอีก

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


304                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

          ENDIF
          GO TOP
          WHILE !EOF()
              AADD(SHOWAR,STR(ORDER))
              SKIP
          END
          @ 2,69,24,79 BOX "+-+|+-+|"
          IND := ACHOICE(3,70,23,78,SHOWAR)
          IF IND > 0 // ปองกันการใชปุม ESC
             _ORDER = VAL(SHOWAR[IND])
          ENDIF
          IF OLDALIAS != "ORDER" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
             CLOSE             // ในจุดที่ไมไดเปดแฟมนี้
          ENDIF
          RESTORE SCREEN FROM SCR
     OTHERWISE
          X = ALERTNEW("ขอโทษ ไมไดเตรียมสวนสนับสนุน ส ําหรับจุดนี้",{"OK."})
  ENDCASE
  IF LEN(OLDALIAS) > 0
     SELE &OLDALIAS // กลับสูพื้นที่ท ํางานเดิม
  ENDIF
RETURN
: ตัวอยางที่ 6.8
&   ผลลัพธ
// สอบถามการเลิกการท ํางาน หากเลิกการทํ างานจะออกเสียง 3 เสียง
PROCEDURE QUITPROC
   Q =  ALERTNEW("สอบถามการเลิกการท ํางาน",{"เลิกการทํ างาน","ยังไมเลิกงาน"})
   IF Q != 2
      FOR I = 1 TO 3
          TONE(50 * I,5-I)
      NEXT
      QUIT
   ENDIF
RETURN

6.4.2 โปรแกรมตาง ๆ ใน SALE1.PRG
โปรแกรมใน SALE1.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 3 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

งานขาย เชน โปรแกรมขายดวยตัวแปรชุดเดียว หลายชุด หรืออาเรย   ทั้ง 3 โปรแกรมนี้เปนโปรแกรมขายทั้งหมด  แต
แตกตางกนัทีห่ลกัการเขียนโปรแกรม เพราะโปรแกรมขายดวยตัวแปรชุดเดียวจะมุงปรับปรุงขอมูลโดยไมมีการ
ค ํานวณยอดตาง ๆ ใหชัดเจน  สวนโปรแกรมขายดวยตัวแปรหลายชุดจะมีตัวแปรมารับผลการคํ านวณตาง ๆ   และ
โปรแกรมขายดวยตัวแปรอาเรยจะท ําการเก็บขอมูลลงอาเรย  สามารถพิมพใบเสร็จโดยอานขอมูลจากตัวแปรอาเรย
มาพิมพได

ระบบงานขายเปนระบบที่ส ําคัญยิ่ง เพราะระบบนี้ถูกพัฒนาขึ้นเพื่อเก็บประวัติการขาย และโปรแกรมใน
ระบบนีจ้ะบนัทึกขอมูลการขายเขาไปเก็บในแฟมสั่งซื้อ ซึ่งเปนขอมูลหลักของระบบ และท ําใหนํ าขอมูลไปปรับปรุง
น ําไปใช และประมวลผลในงานของระบบอื่นไดตอไป

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE1.PRG ทั้งหมด 3 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.9 ถึง ตัวอยางที่ 6.11 ( R0101 - R0103 )
: ตัวอยางที่ 6.9
&   ผลลัพธ
// โปรแกรมรบัขอมูลการขาย โดยใชตัวแปรชุดเดียว คือชุดที่เกี่ยวของกับการปรับปรุง
// หนาที่ของโปรแกรมจะบันทึกขอมูลในแฟมใบสั่งสินคา และใบสั่งสินคาละเอียด
// แลวไปลดยอดสินคาในแฟมสินคา ท ําใหตรวจสอบสินคาคงคลังในภายหลังได
// โดยโปรแกรมจะรับขอมูลทีละรายการ พิมพผลการค ํานวณของรายการ แลวบันทึกขอมูล
PROCEDURE S0101
   _ORDER := _CUST := _EMPL := _SHIPPER := _FREIGHT := 0
   _SHIPADDR = SPACE(40)
   _DATEORD = DATE()
   _DATESHIP = DATE() + 7
   _DATEREQ = CTOD('')
   // ========================
   USE ORDER
   @ 2,5 SAY "ปอนเลขที่ใบสั่งสินคา : " GET _ORDER PICT "99999"
   @ 3,5 SAY "รหัสลูกคา : " GET _CUST PICT "9999"
   @ 4,5 SAY "รหสัพนักงาน : " GET _EMPL PICT "9999"
   @ 5,5 SAY "สถานที่จัดสง : " GET _SHIPADDR
   @ 6,5 SAY "วันที่สั่ง : " GET _DATEORD
   @ 7,5 SAY "วนัที่ตองการ : " GET _DATEREQ
   @ 8,5 SAY "วันที่สง : " GET _DATESHIP

   @ 9,5 SAY "รหัสบริษัทรับขนสง : " GET _SHIPPER PICT "99"
   @ 10,5 SAY "คาขนสง : " GET _FREIGHT PICT "9999.99"
   READ
   APPEND BLANK  // เก็บลงแฟมสั่งสินคา (ORDER.DBF)
   REPLACE ORDER WITH _ORDER,CUST WITH _CUST,EMPL WITH _EMPL,;
           SHIPADDR WITH _SHIPADDR,DATEORD WITH _DATEORD,;
           DATEREQ WITH _DATEREQ,DATESHIP WITH _DATESHIP,;
           SHIPPER WITH _SHIPPER,FREIGHT WITH _FREIGHT
   _SUM = 0
   LN = 11
   DO WHILE LASTKEY() != 27
      USE ORDERD
      _PRO := _OPRICE := _OQUANT := _ODISC := 0
      @ LN,5 SAY "สนิคา" GET _PRO PICT "9999"
      @ LN,18 SAY "ราคาขาย" GET _OPRICE PICT "9999.99"
      @ LN,35 SAY "ปรมิาณ" GET _OQUANT PICT "9999"
      @ LN,52 SAY "สวนลด" GET _ODISC PICT "999.99"
      READ
      IF _PRO != 0
         _SUM = _SUM + (_OPRICE * _OQUANT) - _ODISC
         @ LN, 68 SAY _SUM PICT "9,999,999.99"
         APPEND BLANK    // เกบ็ลงแฟมสั่งสินคาละเอียด (ORDERD.DBF)
         REPLACE ORDER WITH _ORDER,PRO WITH _PRO,OPRICE WITH _OPRICE ,;
            OQUANT WITH _OQUANT,ODISC WITH _ODISC
         // สวนที่ใชลดปริมาณในคลังสินคา
         USE PRO INDEX PRO1
         SEEK _PRO
         IF FOUND()
            REPLACE PROREST WITH PROREST - _OQUANT
         ENDIF
      ELSE
         @ LN+1,68 SAY _SUM PICT "9,999,999.99"
         KEYBOARD CHR(27)

         INKEY()
      ENDIF
      LN++
   ENDDO
   INKEY(3)
RETURN
: ตัวอยางที่ 6.10
&   ผลลัพธ
// โปรแกรมรับขอมูลการขาย โดยใชตัวแปรหลายชุด เพื่อคํ านวณในหลายหนาที่
// หนาที่หลักของโปรแกรมคือ บันทึกขอมูลในแฟมใบสั่งสินคา และใบสั่งสินคาละเอียด
// แลวไปลดยอดสินคาในแฟมสินคา ท ําใหตรวจสอบสินคาคงคลังในภายหลังได
// โดยโปรแกรมจะรับขอมูลทีละรายการ ตรวจจํ านวนในคลังสินคา แลวปรับปรุงขอมูล
// สดุทายแสดงผลรวมเงินของการขายสินคาแตละรายการ ในบรรทัดลางสุด
// แลวสอบถามความตองการพิมพใบเสร็จ ทางเครื่องพิมพหรือจอภาพ
PROCEDURE S0102
   _ORDER := _CUST := _EMPL := _SHIPPER := _FREIGHT := 0
   _SHIPADDR = SPACE(40)  ;    _DATEORD = DATE()
   _DATESHIP = DATE() + 7  ;
   _DATEREQ = CTOD('')
   // ========================
   USE ORDER
   @  2,5 SAY "เลขที่ใบสั่งสินคา :" GET _ORDER PICT "99999"
   @  3,5 SAY "     รหัสลูกคา :" GET _CUST PICT "9999"
   @  4,5 SAY "   รหสัพนักงาน :" GET _EMPL PICT "9999"
   @  5,5 SAY "   สถานที่จัดสง :" GET _SHIPADDR
   @  6,5 SAY "       วันที่สั่ง :" GET _DATEORD
   @  7,5 SAY "   วนัที่ตองการ :" GET _DATEREQ
   @  8,5 SAY "       วันที่สง :" GET _DATESHIP
   @  9,5 SAY " รหัสบริษัทรับขน :" GET _SHIPPER PICT "99"
   @ 10,5 SAY "      คาขนสง :" GET _FREIGHT PICT "9999.99"
   READ
   APPEND BLANK  // เก็บลงแฟมสั่งสินคา (ORDER.DBF)
   REPLACE ORDER WITH _ORDER,CUST WITH _CUST,EMPL WITH _EMPL,;
           SHIPADDR WITH _SHIPADDR,DATEORD WITH _DATEORD,;

           DATEREQ WITH _DATEREQ,DATESHIP WITH _DATESHIP,;
           SHIPPER WITH _SHIPPER,FREIGHT WITH _FREIGHT
   _SUM = 0
   @ 11,10 SAY "รหัสสินคา           ราคา     ปริมาณ  สวนลด  รวมเงิน"
   LN = 12
   DO WHILE LASTKEY() != 27
      _PRO := _OPRICE := _OQUANT := _ODISC := 0
      USE PRO INDEX PRO1
      WHILE .T. // ปอนรหัสสินคาจนกวาจะพบ และมีพอขายใหลูกคา
         @ LN,10 GET _PRO PICT "9999"
         @ LN,35 GET _OPRICE PICT "9999.99"
         @ LN,45 GET _OQUANT PICT "9999"
         @ LN,55 GET _ODISC PICT "999.99"
         READ
         SEEK _PRO
         IF FOUND()
            @ LN,18 SAY LEFT(PRONAME,15)
            IF PROREST >= _OQUANT
               EXIT
            ELSE
               @ LN,18 SAY "ปริมาณมีไมพอขาย"
            ENDIF
         ELSE
            @ LN,18 SAY "ไมพบสินคา"
         ENDIF
         IF _PRO =0
            EXIT
         ENDIF
      END
      IF _PRO != 0
         _SUM = _SUM + (_OPRICE * _OQUANT) - _ODISC
         USE  ORDERD
            APPEND BLANK    // เกบ็ลงแฟมสั่งสินคาละเอียด (ORDERD.DBF)

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                บทที่  6  :  กรณีศึกษาระบบงานขาย และสินคาคงคลัง                                          309

            REPLACE ORDER WITH _ORDER,PRO WITH _PRO,OPRICE WITH _OPRICE ,;
              OQUANT WITH _OQUANT,ODISC WITH _ODISC
        USE PRO INDEX PRO1   // สวนที่ใชลดปริมาณในคลังสินคา
            SEEK _PRO
            IF FOUND()
               REPLACE PROREST WITH PROREST - _OQUANT
            ENDIF
         @ LN,65 SAY STR((_OPRICE * _OQUANT) - _ODISC)
      ELSE
         @ LN,1 CLEAR TO LN,78
         @ LN,65 SAY _SUM PICT "9,999,999.99"
         INKEY(3)
         KEYBOARD CHR(27)
         INKEY()
      ENDIF
      LN++
   ENDDO
   // การสั่งพิมพใบเสร็จ
   IF _SUM > 0
      GASKPRT = ASKPRT()
      IF GASKPRT = 2
         SET DEVICE TO PRINTER
      ENDIF
         CLS
         SELE 1 ; USE ORDERD
         SET FILTER TO ORDER = _ORDER
         GO TOP
         @ 2,10 SAY "ใบสัง่ซื้อเลขที"่ + STR(_ORDER)
         @ 3,10 SAY "======================="
         SELE 2 ; USE PRO INDEX PRO1  NEW
         LN = 4 ; _SUM = 0
         DO WHILE !EOF() // แฟมหลักในลูปนี้คือ แฟมใบสั่งซื้อละเอียด
            SELE 1  ; _PRO = PRO

            SELE 2  ; SEEK _PRO // แฟมสินคา
               IF FOUND()
                  @ LN,10 SAY PRO
                  @ LN,20 SAY LEFT(PRONAME,18)
                ENDIF
            SELE 1   // แฟมใบสั่งซื้อละเอียด
               @ LN,36 SAY OPRICE
               @ LN,44 SAY OQUANT
               @ LN,52 SAY ODISC
               @ LN,60 SAY (OPRICE * OQUANT) - ODISC
               _SUM = _SUM + (OPRICE *OQUANT) - ODISC
               LN++
            SKIP
         ENDDO
      @ LN,60 SAY _SUM
      IF GASKPRT = 2
         EJECT
      ENDIF
      SET DEVICE TO SCREEN
      WAIT "กดปุมใด ๆ เพื่อท ํางานตอ"
      CLOSE ALL
   ENDIF
   // สอบถามการปอนใบสั่งซื้อสินคารายการใหม
   ASKLOOP = ALERTNEW("สอบถามการปอนใบสั่งซื้อรายการใหม",;
            {"ปอนใบสั่งซื้อใหม","เลิกงานสั่งซื้อ"})
   IF ASKLOOP = 1
      CLS
      DO S0102
   ENDIF
RETURN
: ตัวอยางที่ 6.11
&   ผลลัพธ
// โปรแกรมรับขอมูลการขาย โดยใชตัวแปรอาเรย
// หนาที่หลักของโปรแกรมคือ บันทึกขอมูลในแฟมใบสั่งสินคา และใบสั่งสินคาละเอียด

// แลวไปลดยอดสินคาในแฟมสินคา ท ําใหตรวจสอบสินคาคงคลังในภายหลังได
// โดยโปรแกรมจะรับขอมูลทีละรายการ ตรวจจํ านวนในคลังสินคา แลวปรับปรุงขอมูล
// สดุทายแสดงผลรวมเงินของการขายสินคาแตละรายการ ในบรรทัดลางสุด
// แลวสอบถามความตองการพิมพใบเสร็จ ทางเครื่องพิมพหรือจอภาพ
// ในการพมิพขอมูลจะอานขอมูลจากอาเรยที่เก็บไวมาพิมพ แทนการอานจากแฟมโดยตรง
PROCEDURE S0103
   ASKLOOP = 1
   DO WHILE ASKLOOP = 1
      CLS
      _ORDER := _CUST := _EMPL := _SHIPPER := _FREIGHT := 0
      _SHIPADDR = SPACE(40)
      _DATEORD = DATE()
      _DATESHIP = DATE() + 7
      _DATEREQ = CTOD('')
      // ========================
      USE ORDER
      @  2,5 SAY "เลขที่ใบสั่งสินคา :" GET _ORDER PICT "99999"
      @  3,5 SAY "     รหัสลูกคา :" GET _CUST PICT "9999"
      @  4,5 SAY "   รหัสพนักงาน :" GET _EMPL PICT "9999"
      @  5,5 SAY "   สถานที่จัดสง :" GET _SHIPADDR
      @  6,5 SAY "       วันที่สั่ง :" GET _DATEORD
      @  7,5 SAY "   วนัที่ตองการ :" GET _DATEREQ
      @  8,5 SAY "       วันที่สง :" GET _DATESHIP
      @  9,5 SAY " รหัสบริษัทรับขน :" GET _SHIPPER PICT "99"
      @ 10,5 SAY "      คาขนสง :" GET _FREIGHT PICT "9999.99"
      READ
      IF UPDATED()
         APPEND BLANK  // เก็บลงแฟมสั่งสินคา (ORDER.DBF)
         REPLACE ORDER WITH _ORDER,CUST WITH _CUST,EMPL WITH _EMPL,;
              SHIPADDR WITH _SHIPADDR,DATEORD WITH _DATEORD,;
              DATEREQ WITH _DATEREQ,DATESHIP WITH _DATESHIP,;
              SHIPPER WITH _SHIPPER,FREIGHT WITH _FREIGHT
      ELSE

         RETURN
      ENDIF
      _SUM = 0
      LN = 12 ; J = 1
      _PRO := ARRAY(15)
      _OPRICE := ARRAY(15)
      _OQUANT := ARRAY(15)
      _ODISC := ARRAY(15)
      _MONEY := ARRAY(15)
      @ 11,10 SAY "รหัสสินคา           ราคา     ปริมาณ  สวนลด  รวมเงิน"
      DO WHILE LASTKEY() != 27
         USE PRO INDEX PRO1
         _PRO[J] = 1
         DO WHILE _PRO[J] != 0
            _PRO[J] := _OPRICE[J] := _OQUANT[J] := _ODISC[J] := 0
            @ LN,10 GET _PRO[J] PICT "9999"
            @ LN,35 GET _OPRICE[J] PICT "9999.99"
            @ LN,45 GET _OQUANT[J] PICT "9999"
            @ LN,55 GET _ODISC[J] PICT "999.99"
            READ
            SEEK _PRO[J]
            IF FOUND()
               @ LN,18 SAY LEFT(PRONAME,15)
               IF PROREST >= _OQUANT[J]  // ปรับยอดในคลังสินคา
                  REPLACE PROREST WITH PROREST - _OQUANT[J]
                  _MONEY[J] = _OPRICE[J]*_OQUANT[J]-_ODISC[J]
                  @ LN,65 SAY STR(_MONEY[J])
                  EXIT
               ENDIF
            ENDIF
         ENDDO
         IF _PRO[J] != 0
            _SUM = _SUM + _MONEY[J]

            USE  ORDERD
            APPEND BLANK    // เกบ็ลงแฟมสั่งสินคาละเอียด (ORDERD.DBF)
            REPLACE ORDER  WITH _ORDER,;
                    PRO    WITH _PRO[J],;
                    OPRICE WITH _OPRICE[J],;
                    OQUANT WITH _OQUANT[J],;
                    ODISC  WITH _ODISC[J]
         ELSE
            @ LN,1 CLEAR TO LN,78
            @ LN,65 SAY _SUM PICT "9,999,999.99"
            INKEY(3) ;  KEYBOARD CHR(27) ;  INKEY()
         ENDIF
         LN++  ; J++
      ENDDO
      // การสั่งพิมพใบเสร็จ
      IF _SUM > 0
         ALLPRO = J - 2
         GASKPRT = ASKPRT()
         IF GASKPRT = 2
            SET DEVICE TO PRINTER
         ENDIF
         CLS
         @ 2,10 SAY "ใบสัง่ซื้อเลขที"่ + STR(_ORDER)
         @ 3,10 SAY "======================="
         USE PRO INDEX PRO1
         LN = 4
         FOR I = 1 TO ALLPRO
             SEEK _PRO[I] // แฟมสินคา
             IF FOUND()
                @ LN,10 SAY PRO
                @ LN,20 SAY LEFT(PRONAME,18)
             ENDIF
             @ LN,36 SAY _OPRICE[I]

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


314                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

             @ LN,44 SAY _OQUANT[I]
             @ LN,52 SAY _ODISC[I]
             @ LN,60 SAY _MONEY[I]
             LN++
          NEXT
          @ LN,60 SAY _SUM
          IF GASKPRT = 2 ; EJECT ; ENDIF
          SET DEVICE TO SCREEN
          WAIT "กดปุมใด ๆ เพื่อท ํางานตอ"
          CLOSE ALL
       ENDIF
       ASKLOOP = ALERTNEW("สอบถามการปอนใบสั่งซื้อรายการใหม",;
               {"ปอนใบสั่งซื้อใหม","เลิกงานสั่งซื้อ"})
   ENDDO
RETURN

6.4.3 โปรแกรมตาง ๆ ใน SALE2.PRG
โปรแกรมใน SALE2.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 5 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

ปรบัปรงุ เชน โปรแกรมปรับปรุงใบสั่งสินคา  โปรแกรมเพิ่มเงินเดือนใหพนักงาน โปรแกรมยกเลิกใบสั่งสินคา  หรือ
การปรับปรุงแฟมตาง ๆ เปนตน

ระบบปรบัปรงุเปนระบบที่ชวยแกปญหาตาง ๆ เชนปอนขอมูลการสั่งซื้อผิด หรือลูกคายกเลิกใบสั่งซื้อรวม
ถงึการเพิม่เงนิเดือนใหพนักงาน และสามารถแกไขขอมูลในแฟมทุกแฟมไดอยางงาย ๆ

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE2.PRG ทั้งหมด 5 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.12 ถึง ตัวอยางที่ 6.16 ( R0201 - R0205 )
: ตัวอยางที่ 6.12
&   ผลลัพธ
// ปรบัปรุงใบสั่งสินคา โดยรับรหัสใบสั่ง แลวน ําขอมูลในใบสั่งสินคามาใหแกไข
// เมือ่แกไขขอมูลในใบสงสินคาแลว จะนํ าขอมูลในใบสั่งสินคาละเอียดเก็บลงอาเรย
// แลวน ําขอมูลในอาเรยมาใหปรับปรุง เมื่อปรับปรุงแลวจะเขียนทับลงไปในเรคอรดเดิม
PROCEDURE S0201
   _ORDER := 0
   DO WHILE LASTKEY() != 27
      @ 1,5 SAY "เลขที่ใบสั่งซื้อ : " GET _ORDER PICT "99999"
      READ

      USE ORDER
      LOCATE FOR ORDER = _ORDER
      IF FOUND()
         @ 2,5 SAY "รหัสลูกคา : "  GET CUST
         @ 3,5 SAY "รหัสพนักงาน : " GET EMPL
         @ 4,5 SAY "สถานที่จัดสง : " GET SHIPADDR
         @ 5,5 SAY "วนัที่สั่งสินคา : " GET DATEORD
         @ 6,5 SAY "วนัทีต่องการสินคา : "GET DATEREQ
         @ 7,5 SAY "วนัที่จัดสงสินคา : " GET DATESHIP
         @ 8,5 SAY "บริษัทจัดสง : " GET SHIPPER
         @ 9,5 SAY "คาขนสง : " GET FREIGHT
         READ
         CLOSE // ฟลดจะไดรับการปรับปรุงทันท ีโดยไมตอง REPLACE
         SELE 1 ;  USE ORDERD
            SET FILTER TO ORDER = _ORDER
            COUNT TO CNTALL FOR ORDER = _ORDER
            GO TOP
            _PRO := ARRAY(15)
            _OPRICE := ARRAY(15)
            _OQUANT := ARRAY(15)
            _ODISC := ARRAY(15)
            FOR I := 1 TO CNTALL  // เก็บคาลงอาเรย
               _PRO[I] = PRO
               _OPRICE[I] = OPRICE
               _OQUANT[I] = OQUANT
               _ODISC[I] = ODISC
               SKIP
            NEXT
            FOR J := 1 TO CNTALL // รับคาปรับปรุง
                @ J+10, 10 GET _PRO[J]
                @ J+10, 40 GET _OPRICE[J]
                @ J+10, 50 GET _OQUANT[J]
                @ J+10, 57 GET _ODISC[J]

                SELE 2 ; USE PRO INDEX PRO1
                SEEK _PRO[J]
                IF FOUND()
                   @ J+10,18 SAY LEFT(PRONAME,20)
                   @ J+10,65 SAY _OPRICE[J] * _OQUANT[J] - _ODISC[J]
                ENDIF
            NEXT
            READ
            IF UPDATED()  // ตรวจสอบวาปรับปรุงหรือไม
               SELE 1
               GO TOP
               FOR I := 1 TO CNTALL
                  REPLACE PRO WITH _PRO[I],OPRICE WITH _OPRICE[I],;
                         OQUANT WITH _OQUANT[I],ODISC WITH _ODISC[I]
                  SKIP
               NEXT
            ENDIF
         CLOSE ALL
         EXIT // ถาปรับปรุงเสร็จแลว จะเลิกโปรแกรมนี้ทันที
      ENDIF
   ENDDO
RETURN
: ตัวอยางที่ 6.13
&   ผลลัพธ
// ยกเลกิใบสั่งสินคา โดยรับเลขที่ใบสั่งสินคาที่ตองการลบ แลวคนหาเรคอรดที่ตรง
// กบัในแฟมใบสั่งซื้อ และใบสั่งซื้อละเอียด
// เมือ่พบจะลงเรคอรดจากทั้ง 2 แฟมที่มีเลขที่ใบสั่งสินคาตรงกับที่ตองการ
PROCEDURE S0202
   _ORDER = 0
   USE ORDER
   DO WHILE .T.
      @ 5,5 SAY "เลขทีใ่บสั่งซื้อที่ตองการลบ : " GET _ORDER PICT "99999"
      READ
      LOCATE FOR _ORDER = ORDER

      IF FOUND() // เหน็เพยีงเรคอรดเดียวในตารางของ DBEDIT
         DBEDIT(5,10,23,70,{"ORDER","CUST","EMPL"})
         IF ASKUPD() = 1
            DELETE
            PACK
            USE ORDERD
            DELETE ALL FOR ORDER = _ORDER
            PACK
         ENDIF
         EXIT
      ENDIF
   ENDDO
RETURN
: ตัวอยางที่ 6.14
&   ผลลัพธ
// ยกเลกิสินคาในใบสั่งสินคาบางรายการ โดยรับเลขที่ใบสั่งสินคา
// แลวเลอืกเรคอรดที่เกี่ยวของจากแฟมใบสั่งสินคาละเอียดมาใหปรับปรุงโดยงาย
// โดยการแกไขขอมูลใชฟงกชัน BROWSE หลังจากที่เลือกขอมูลที่ตรงกับเลขที่ใบสั่ง
PROCEDURE S0203
   _ORDER = 0
   USE ORDERD
   DO WHILE .T.
      @ 5,5 SAY "เลขทีใ่บสั่งซื้อที่ตองการลบสินคา : " GET _ORDER PICT "99999"
      READ
      LOCATE FOR _ORDER = ORDER
      IF FOUND()
         SET FILTER TO ORDER = _ORDER
         BROWSE(5,10,20,70)
         PACK
         EXIT
      ENDIF
   ENDDO
RETURN

: ตัวอยางที่ 6.15
&   ผลลัพธ
// ปรบัปรงุเงินเดือนพนักงานทุกคน โดยปอนเปอรเซนตของการปรับปรุง
// แลวจงึใชคํ าสั่งปรับปรุงขอมูลพรอมกันใหพนักงานทุกคน
PROCEDURE S0204
   _ADDING = 0
   @ 5,5 SAY "จ ํานวนเปอรเซนตที่เพิ่มเงินเดือนใหพนักงาน : " GET _ADDING
   READ
   IF ASKUPD() = 1
      USE EMPL
      REPLACE ALL EMPLSAL WITH EMPLSAL + (EMPLSAL * _ADDING / 100)
      CLOSE
   ENDIF
RETURN
: ตัวอยางที่ 6.16
&   ผลลัพธ
// ปรบัปรงุแฟมตาง ๆ อยางงาย โดยเลือกแฟมจากตัวเลือก แลวเปดแฟม
// จากนัน้ใชฟงกชัน BROWSE ชวยใหการปรับปรุงขอมูลงายขึ้นมาก
PROCEDURE S0205
   OMENU = ALERTNEW("เลอืกปรับปรุงแฟมตามตองการ",;
          {"แฟมสั่งสินคา","แฟมสั่งสินคาละเอียด","แฟมสินคา",;
           "แฟมพนักงาน","แฟมลูกคา","แฟมบริษัทรับสงของ",;
           "แฟมประเภทสินคา","แฟมผูผลิต","กลับเมนู"})
   DO CASE
      CASE OMENU = 1 ; USE ORDER
      CASE OMENU = 2 ; USE ORDERD
      CASE OMENU = 3 ; USE PRO
      CASE OMENU = 4 ; USE EMPL
      CASE OMENU = 5 ; USE CUST
      CASE OMENU = 6 ; USE SHIPPER
      CASE OMENU = 7 ; USE CAT
      CASE OMENU = 8 ; USE SUP
   ENDCASE
   IF OMENU >= 1 .AND. OMENU <= 8

      BROWSE(5,5,22,75)
      PACK
   ENDIF
   CLOSE ALL
RETURN

6.4.4 โปรแกรมตาง ๆ ใน SALE3.PRG
โปรแกรมใน SALE3.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 8 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

รายงาน เชน โปรแกรมพิมพรายงานยอดขายของสินคา ตามเงื่อนไขตาง ๆ   โปรแกรมพิมพรายงานใบสั่งสินคาตาม
เงือ่นไขตาง ๆ   และโปรแกรมสรุปใบสั่งซื้อของแตละวัน

ระบบรายงานเปนระบบที่ชวยท ําใหเห็นคุณคา และผลงานจากระบบงานขาย  ท ําใหผูบริหารบริหารสินคา
ในคลงั วางแผนงานหรือนโยบายทางการตลาดไดอยางถูกตอง เพราะรายงานจะแสดงยอดขาย แสดงการรับบริการ
ของลูกคา  เหน็ยอดจ ําหนายของสินคาแตละประเภท  จัดการกับสินคาที่ขายไมได หรือสินคาที่ขายดี ใหขายดียิ่ง ๆ
ขึน้  เพื่อใหบริษัทเจริญรุงเรือง และเติบโตตอไป

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE3.PRG ทั้งหมด 8 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.17 ถึง ตัวอยางที่ 6.24 ( R0301 - R0308 )
: ตัวอยางที่ 6.17
&   ผลลัพธ
// พมิพสรุปยอดขายของสินคา โดยระบุเดือนที่สั่งสินคา
// รายงานแสดงรหัสสินคา และจ ํานวนเงิน ที่ไดทั้งหมด ตามเดือน
PROCEDURE S0301
   USE ORDERD
   SORT ON PRO,ORDER TO TEMPF
   USE TEMPF
   COPY TO ORDERD
   _MONTH = 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   READ
   SELE 1 ; USE ORDER
   SELE 2 ; USE  ORDERD
   _PROO = 0
   _TQUANT = 0

   CHKLINE = 1
   DO WHILE !EOF()
      SELE 2  // แฟมใบสั่งสินคาละเอียด
         _ORDER = ORDER
      SELE 1  // แฟมใบสั่งสินคา
         _FOUNDM = 0
         LOCATE FOR ORDER = _ORDER
         IF FOUND()
            IF MONTH(DATEORD) = _MONTH  // ตรวจเดือนที่ตองการ
               _FOUNDM = 1
            ENDIF
         ENDIF
      SELE 2 // แฟมใบสั่งสินคา
         IF _FOUNDM = 1
            IF PRO != _PROO // ถาตางกลุม
               ? PRO
               _PROO = PRO
               CHKLINE = 0
            ENDIF
            _TQUANT = _TQUANT + OQUANT
         ENDIF
         SKIP
         IF (_PROO != PRO .OR. EOF()) .AND. CHKLINE = 0
            ?? _TQUANT  // ถาเปลี่ยนกลุม
            _TQUANT = 0
            CHKLINE = 1 // แสดงการพิมพบรรทัดแตละบรรทัดสมบูรณ
         ENDIF
   ENDDO
   CLOSE ALL
   INKEY(3)
RETURN

: ตัวอยางที่ 6.18
&   ผลลัพธ
// พมิพสรุปยอดขายของสินคา โดยระบุผูผลิตสินคา
// รายงานแสดงรหัสสินคา ชื่อสินคา ปริมาณขาย และจ ํานวนเงินของสินคา ตามผูผลิต
PROCEDURE S0302
   _SUP = 0
   @ 5,5 SAY "ผูผลิตสินคา : " GET _SUP PICT "999"
   READ
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDERD
   INDEX ON PRO TO ORDERD2
   SET RELATION TO PRO INTO PRO
   _PROO = 0
   _TQUANT = 0
   _TMONEY = 0
   DO WHILE !EOF()
      IF PRO->SUP = _SUP
         IF _PROO != PRO // ถาตางกลุม
            ? PRO,PRO->PRONAME
            _PROO = PRO
         ENDIF
         _TQUANT = _TQUANT + OQUANT
         _TMONEY = _TMONEY + (OPRICE * OQUANT - ODISC)
         SKIP
         IF _PROO != PRO .OR.  EOF()
            ?? _TQUANT,_TMONEY  // ถาเปลี่ยนกลุม
            _TQUANT := _TMONEY := 0
         ENDIF  // ตรวจกลุมของสินคา
      ELSE // ถาไมใชผูผลิตที่ตองการ
         SKIP
      ENDIF // ตรวจผูผลิต
   ENDDO
   CLOSE ALL
   INKEY(3)

RETURN
: ตัวอยางที่ 6.19
&   ผลลัพธ
// พมิพสรุปยอดขายของสินคา โดยระบุผูผลิตสินคา และเดือนที่สั่ง
// รายงานแสดงรหัสสินคา ชื่อสินคา และปริมาณ ตามผูผลิตและเดือน
// และแสดงใหเห็นการเชื่อมแฟมสินคา ใบสั่งสินคา และใบสั่งสินคาละเอียด
PROCEDURE S0303
   _SUP := _MONTH := 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   @ 6,5 SAY "ผูผลิตสินคา : " GET _SUP PICT "999"
   READ
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDER
   SELE 3 ; USE ORDERD
   INDEX ON PRO TO ORDERD2
   SET RELATION TO PRO INTO PRO
   _PROO = 0
   _TQUANT = 0
   CHKLINE = 1
   DO WHILE !EOF()
      _ORDER = ORDER
      SELE 2  // แฟมใบสั่งสินคา
         _FOUNDM = 0
         LOCATE FOR _ORDER = ORDER .AND. MONTH(DATEORD) = _MONTH
         IF FOUND()
            _FOUNDM = 1
         ENDIF
      SELE 3 // แฟมใบสั่งสินคา
         IF _FOUNDM = 1 .AND. PRO->SUP = _SUP
            IF PRO != _PROO  // ถาตางกลุม
               ? PRO,PRO->PRONAME
               _PROO = PRO
               CHKLINE = 0
            ENDIF

            _TQUANT = _TQUANT + OQUANT
         ENDIF
         SKIP
         IF (_PROO != PRO .OR. EOF()) .AND. CHKLINE = 0
            ?? _TQUANT  // ถาเปลี่ยนกลุม
            _TQUANT = 0
            CHKLINE = 1 // แสดงการพิมพบรรทัดแตละบรรทัดสมบูรณ
         ENDIF
   ENDDO
   CLOSE ALL   ;   INKEY(3)
RETURN
: ตัวอยางที่ 6.20
&   ผลลัพธ
// พมิพสรุปยอดขายของประเภทสินคา โดยระบุเดือนที่สั่ง
// รายงานแสดงรหัสใบสั่ง รหัสสินคา ชื่อสินคา และจ ํานวนเงินในใบสั่ง ตามประเภท
// แลวสรุปยอดขายของสินคาแตละประเภท
// และแสดงใหเห็นการเชื่อมแฟมสินคา ใบสั่งสินคา และใบสั่งสินคาละเอียด
PROCEDURE S0304
   _MONTH := 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   READ
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDER
   SELE 3 ; USE ORDERD INDEX ORDERD2
   SET RELATION TO PRO INTO PRO
   SELE 4 ; USE CAT
   _CAT = 0
   _TQUANT = 0
   DO WHILE !EOF()
      _CAT = CAT
      ? CAT,CATNAME
      SELE 3 // แฟมใบสั่งสินคาละเอียด
         SET FILTER TO PRO->CAT = _CAT // จับกลุมใหประเภทสินคา
         GO TOP

         _TQUANT = 0
         DO WHILE !EOF()
            _ORDER = ORDER
            SELE 2   // แฟมใบสั่งสินคา
            LOCATE FOR ORDER = _ORDER .AND. MONTH(DATEORD) = _MONTH
            IF FOUND()
                 SELE 3
                 _TQUANT = _TQUANT + OQUANT
                 ? ORDER,PRO,PRO->PRONAME,OQUANT
            ENDIF
            SELE 3
            SKIP
         ENDDO
      SELE 4 // แฟมประเภทสินคา
      ? "         จํ านวน = ",_TQUANT
      ? "==============================="
      SKIP
   ENDDO
   CLOSE ALL
   INKEY(3)
RETURN

: ตัวอยางที่ 6.21
&   ผลลัพธ
// พมิพสรุปยอดขายใบสั่งสินคาแตละใบ โดยระบุวันที่
// รายงานแสดงล ําดับ รหัสใบสั่ง และยอดขายของใบสั่งสินคา ตามวันที่
// โดยการหายอดขายจะใชประโยชนจากตัวแปรอาเรยมาเก็บขอมูล
// แลวจงึน ําขอมูลจากอาเรยมาพิมพอีกครั้งหนึ่ง
PROCEDURE S0305
   SELE 1 ; USE ORDERD
   SELE 2 ; USE ORDER
   DO WHILE LASTKEY() != 27
      _DATEORD = CTOD('')
      @ 5,5 SAY "วนัที่ที่มีการสั่งสินคา :" GET _DATEORD

      READ
      SELE 2
      COUNT TO CNTORDER FOR DATEORD = _DATEORD
      IF CNTORDER > 0
         @ 6,5 SAY "จ ํานวนใบสั่งซื้อ :" + STR(CNTORDER)
         SET FILTER TO DATEORD = _DATEORD
         GO TOP
         ARORDER = {}
         DO WHILE !EOF()
            AADD(ARORDER,ORDER)
            SKIP
         ENDDO
         ARMONEY = ARRAY(CNTORDER)
         AFILL(ARMONEY,0)
         SELE 1  // เริม่หาจ ํานวนเงินของใบสั่งแตละใบ ในแฟมใบสั่งสินคาละเอียด
         FOR I = 1 TO CNTORDER
             SET FILTER TO ORDER = ARORDER[I]
             GO TOP
             _MONEY = 0
             DO WHILE ! EOF()
                _MONEY = _MONEY + OPRICE * OQUANT - ODISC
                SKIP
             ENDDO
             ARMONEY[I] = _MONEY
          NEXT
          @ 8,5 SAY "ล ําดับ รหัสใบสั่งสินคา จํ านวนเงิน"
          FOR I = 1 TO CNTORDER
              @ 8+I,5 SAY STR(I,4)+STR(ARORDER[I],6)+STR(ARMONEY[I],9)
          NEXT
      ELSE
          @ 8,5 SAY "NOT FOUND"
      ENDIF
      INKEY(0)

      @ 8,5 CLEAR TO 20,70
   ENDDO
RETURN
: ตัวอยางที่ 6.22
&   ผลลัพธ
// พมิพสรุปยอดขายใบสั่งสินคาแตละใบ โดยระบุรหัสพนักงานขาย
// รายงานแสดงล ําดับ รหัสใบสั่ง และยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย
// ในการปอนรหัสพนักงานขาย ตองปอนจนกวาจะมีรหัสนั้นในแฟมจริง ๆ จึงจะยอมรับ
// โดยการหายอดขายจะใชประโยชนจากตัวแปรอาเรยมาเก็บขอมูล
// แลวจงึน ําขอมูลจากอาเรยมาพิมพอีกครั้งหนึ่ง ซึ่งเลือกปลายทางของผลลัพธได
PROCEDURE S0306
   SELE 1 ; USE EMPL
   SELE 2 ; USE ORDERD
   SELE 3 ; USE ORDER
   DO WHILE LASTKEY() != 27
      _EMPL = 0
      SELE 1
         DO WHILE LASTKEY() != 27
            @ 5,5 SAY "รหสัพนักงาน :" GET _EMPL PICT "9999"
            READ
            LOCATE FOR EMPL = _EMPL
            IF FOUND()
               @ 7,5 SAY EMPLNAME
               EXIT
            ENDIF
         ENDDO
      SELE 3
      COUNT TO CNTEMPL FOR EMPL = _EMPL
      IF CNTEMPL > 0
         GASKPRT = ASKPRT()    // สอบถามการนํ าออกเครื่องพิมพ
         IF GASKPRT = 2
            SET DEVICE TO PRINTER
            @ 5,5 SAY "รหสัพนักงาน : "+STR(_EMPL)
         ENDIF

         @ 6,5 SAY "จ ํานวนใบสั่งซื้อ :" + STR(CNTEMPL)
         SET FILTER TO EMPL = _EMPL
         GO TOP
         ARORDER = {}
         DO WHILE !EOF()
            AADD(ARORDER,ORDER)
            SKIP
         ENDDO
         ARMONEY = ARRAY(CNTEMPL)
         AFILL(ARMONEY,0)
         _TMONEY = 0
         SELE 2  // เริม่หาจ ํานวนเงินของใบสั่งแตละใบ ในแฟมใบสั่งสินคาละเอียด
         FOR I = 1 TO CNTEMPL
             SET FILTER TO ORDER = ARORDER[I]
             GO TOP
             _MONEY = 0
             DO WHILE ! EOF()
                _MONEY = _MONEY + OPRICE * OQUANT - ODISC
                SKIP
             ENDDO
             ARMONEY[I] = _MONEY
             _TMONEY = _TMONEY + _MONEY
          NEXT
          @ 8,5 SAY "ล ําดับ รหัสใบสั่งสินคา จํ านวนเงิน"
          FOR I = 1 TO CNTEMPL
              @ 8+I,5 SAY STR(I,4)+STR(ARORDER[I],6)+STR(ARMONEY[I],9)
          NEXT
          @ 9+CNTEMPL,5 SAY "รวมเงิน : "+STR(_TMONEY)
          IF GASKPRT = 2
             EJECT
          ENDIF
          SET DEVICE TO SCREEN
      ELSE

          @ 8,5 SAY "NOT FOUND"
      ENDIF
      INKEY(0)
      @ 7,5 CLEAR TO 20,70
   ENDDO
RETURN
: ตัวอยางที่ 6.23
&   ผลลัพธ
// พมิพสรุปยอดขายสินคา โดยระบุประเภท
// รายงานแสดงยอดขายของสินคาแตละรายการในแฟมใบสั่งสินคาละเอียด ตามประเภท
// โดยแสดงล ําดับ เลขที่ใบสั่งซื้อ ชื่อสินคา และจ ํานวนเงิน บรรทัดสุดทายแสดงยอดรวม
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE S0307
   _CAT := 0
   @ 5,5 SAY "ประเภทสินคา : " GET _CAT PICT "99"
   READ
   USE PRO INDEX PRO1
   USE ORDERD NEW
   SET RELATION TO PRO INTO PRO
   SET FILTER TO PRO->CAT = _CAT
   COUNT TO CNTCAT
   GO TOP
   IF CNTCAT > 0
      GASKPRT = ASKPRT()
      IF GASKPRT = 2
         SET PRINT ON
      ENDIF
      SETPOS(7,5);  ? _CAT," : ",CNTCAT," RECORDS"
      I = 1  ;      _TMONEY = 0
      DO WHILE !EOF()
         _MONEY = OPRICE * OQUANT - ODISC
         _TMONEY = _TMONEY + _MONEY
         SETPOS(I+7,5); ? I,ORDER,LEFT(PRO->PRONAME,20),_MONEY
         SKIP

         I++
      ENDDO
      SETPOS(I+7,5); ? "เปนเงินทั้งหมด = ",_TMONEY
      IF GASKPRT = 2
         EJECT
         SET PRINT OFF
      ENDIF
   ELSE
      @ 6,5 SAY "NOT FOUND"
   ENDIF
   INKEY(5)
   @ 6,5 CLEAR TO 25,75
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.24
&   ผลลัพธ
// พมิพรายการขายแตละรายการ โดยระบุวันที่
// รายงานแสดงล ําดับ รหัสสินคา ราคา ปริมาณ และสวนลด ตามวันที่
// และไดผลลัพธทางจอภาพเทานั้น
PROCEDURE S0308
   _DATEORD := CTOD('')
   @ 5,5 SAY "วนัที่ลูกคาสั่ง : " GET _DATEORD
   READ
   SELE 1 ; USE ORDERD
   INDEX ON STR(ORDER)+STR(PRO) TO ORDERD1
   SELE 2 ; USE ORDER
      SET FILTER TO  DATEORD = _DATEORD
      COUNT TO CNTORD
   IF CNTORD > 0
      GO TOP ; SETPOS(7,1)
      DO WHILE !EOF()
         _ORDER = ORDER
         SELE 1
         SET FILTER TO ORDER = _ORDER

         ? "*",_ORDER
         GO TOP ; I = 1
         DO WHILE !EOF()
            ? I,PRO,OPRICE,OQUANT,ODISC
            SKIP ; I++
         ENDDO
         ? "==============================="
         SELE 2 ; SKIP
      ENDDO
   ELSE
      @ 8,5 SAY "NOT FOUND"
   ENDIF
   INKEY(3)   ;   CLOSE ALL
RETURN

6.4.5 โปรแกรมตาง ๆ ใน SALE4.PRG
โปรแกรมใน SALE4.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 7 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

สนิคาคงคลงั เชน โปรแกรมพิมพรายงานสินคาที่ถึงจุดสั่งซื้อ โปรแกรมเพิ่มจํ านวนสินคา โปรแกรมเพิ่มรายการสินคา
ใหมในคลงั  โปรแกรมรายงานเกี่ยวกับจํ านวนสินคาแบบตาง ๆ

ระบบสนิคาคงคลงัเปนระบบที่ชวยท ําใหบริษัทมีสินคาจํ าหนายตลอดเวลาเมื่อลูกคาถาม และไมมีสินคาใน
คลงัมากเกนิไป เพราะการสั่งสินคามาก ๆ ท ําใหเงินทุนจมอยูในคลัง  ซึ่งเปนการบริหารเงินที่ไมดีนัก  ดังนั้นในระบบ
นีจ้งึชวยผูบรหิารใจการตัดสินใจซื้อสินคาเพิ่มเมื่อใด เทาใด และจากใคร

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE4.PRG ทั้งหมด 7 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.24 ถึง ตัวอยางที่ 6.30 ( R0401 - R0407 )
: ตัวอยางที่ 6.24
&   ผลลัพธ
// เพิม่ยอดสนิคาเขามาในคลังสินคา ของสินคาแตละรายการ
// มกักระท ําเมื่อสินคาในคลังหมด แลวซื้อเพิ่ม จึงตองเพิ่มจํ านวนสินคาในแฟม
// โดยเพิม่ยอดสินคาไดทีละรายการ หลายรายการ จนกวาจะกดปุม ESC
PROCEDURE S0401
   USE PRO INDEX PRO1
   LN = 5
   DO WHILE LASTKEY() != 27
      _PRO := _PROREST := 0
      @ LN,10 SAY "รหัสสินคา : " GET _PRO PICT "9999"   ;     READ
      SEEK _PRO
      IF FOUND()
         @ LN,30 SAY LEFT(PRONAME,15)
         @ LN,50 SAY "เพิม่จ ํานวนสินคาอีก :" GET _PROREST PICT "999"
         READ
         IF ASKUPD() = 1
            REPLACE PROREST WITH PROREST + _PROREST,PROONORD WITH 0
         ENDIF
         LN++
      ELSE
         @ LN,50 SAY "NOT FOUND"
      ENDIF
   ENDDO
RETURN

: ตัวอยางที่ 6.25
&   ผลลัพธ
// เพิม่ขอมลูสนิคาใหมเขาไปในแฟมสินคา เมื่อมีสินคาชนิดใหมที่ไมเคยมีจํ าหนาย
// เมือ่รอรับขอมูลจะแสดงรหัสสินคารายการใหมมาใหอัตโนมัติ
// โดยเพิม่รายการสินคาไดทีละรายการ หลายรายการจนกวาจะกดปุม ESC
PROCEDURE S0402
   USE PRO INDEX PRO1
   _PRO = 1
   DO WHILE _PRO != 0 .AND. LASTKEY() !=27
      GO BOTTOM
      _PRO = PRO + 1
      _SUP = 0   ;      _CAT = 0
      _PRONAME = SPACE(40)
      _PROUNIT = SPACE(10)
      _PROPRICE := _PROREST := _PROONORD := _REORDER := 0
      @ 10,5 SAY "รหสัสินคา" GET _PRO PICT "9999"
      @ 11,5 SAY "รหัสผูผลิต" GET _SUP PICT "999"
      @ 12,5 SAY "รหัสประเภท" GET _CAT PICT "99"
      @ 13,5 SAY "ชือ่สินคา" GET _PRONAME
      @ 14,5 SAY "ชือ่หนวย" GET _PROUNIT
      @ 15,5 SAY "ราคาทุน" GET _PROPRICE PICT "9999.99"
      @ 16,5 SAY "จ ํานวนที่เหลือ" GET _PROREST PICT "9999"
      @ 17,5 SAY "จ ํานวนที่ก ําลังสั่งเพิ่ม" GET _PROONORD PICT "9999"
      @ 18,5 SAY "จดุสัง่ซื้อ" GET _REORDER PICT "999"
      READ
      IF UPDATED() .AND. ASKUPD() = 1
         APPEND BLANK
         REPLACE PRO WITH _PRO,SUP WITH _SUP,CAT WITH _CAT,;
                 PRONAME WITH _PRONAME,PROUNIT WITH _PROUNIT,;
                 PROPRICE WITH _PROPRICE,PROREST WITH _PROREST,;
                 PROONORD WITH _PROONORD
      ENDIF
   ENDDO
RETURN

: ตัวอยางที่ 6.26
&   ผลลัพธ
// พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
// รายงานแสดงล ําดับ รหัสสินคา ชื่อ จํ านวนเหลือ จํ านวนสั่งเพิ่ม และจุดสั่งซื้อ
PROCEDURE S0403
   USE PRO INDEX PRO1
   REINDEX
   COUNT TO CNTREORDER FOR PROREST < REORDER
   @ 5,5 SAY "จ ํานวนสินคาที่ถึงจุดสั่งซื้อ = " + STR(CNTREORDER)
   IF CNTREORDER > 0
      CLS ; GO TOP ; I = 1
      DO WHILE !EOF()
         IF PROREST < REORDER
            @ ROW()+1,2 SAY I
            @ ROW(),15 SAY PRO
            @ ROW(),25 SAY PRONAME
            @ ROW(),35 SAY PROREST
            @ ROW(),45 SAY PROONORD
            @ ROW(),55 SAY REORDER
            I++
         ENDIF
         SKIP
      ENDDO
   ELSE
      @ 6,5 SAY "ยงัไมมีสินคาชนิดใดถึงจุดสั่งซื้อ"
   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN

: ตัวอยางที่ 6.27
&   ผลลัพธ
// พมิพรายการสินคาที่ ถึงจุดสั่งซื้อตามผูผลิต และยังไมไดสั่งเพิ่ม
// รายงานแสดงล ําดับ รหัสสินคา ชื่อ จํ านวนเหลือ จํ านวนสั่งเพิ่ม จุดสั่งซื้อ
// รายงานหยุดใหดูหนาละ 20 รายการ
PROCEDURE S0404
   _SUP = 0
   @ 5,5 SAY "รหัสผูผลิต : " GET _SUP PICT "999"
   READ
   USE PRO INDEX PRO1
   REINDEX
   CLS  ;  I := 0 ; J := 0
   DO WHILE !EOF()
      IF REORDER > PROREST .AND. SUP = _SUP .AND. PROONORD = 0
         I++ ; J++
         @ J,2  SAY I
         @ J,15 SAY PRO
         @ J,25 SAY PRONAME
         @ J,35 SAY PROREST
         @ J,45 SAY PROONORD
         @ J,55 SAY REORDER
         IF MOD(I,20) = 0
            INKEY(3)
            J := 0
            CLS
         ENDIF
      ENDIF
      SKIP
   ENDDO
   INKEY(3)
   CLOSE ALL
RETURN

: ตัวอยางที่ 6.28
&   ผลลัพธ
// พมิพรายการสินคาที่เหลือไมถึง 10 เปอรเซนตจะถึงจุดสั่งซื้อ
// รายงานแสดงล ําดับ รหัสสินคา ชื่อ จํ านวนเหลือ จํ านวนสั่งเพิ่ม จุดสั่งซื้อ
// รายงานหยุดใหดูหนาละ 20 รายการ
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE S0405
   USE PRO INDEX PRO1
   REINDEX
   COUNT TO CNTREORDER FOR PROREST < REORDER * 1.1
   @ 5,5 SAY "จ ํานวนสินคาใกล ถึงจุดสั่งซื้อ = " + STR(CNTREORDER)
   IF CNTREORDER > 0
      INKEY(1) ; CLS
      GASKPRT = ASKPRT()
      IF GASKPRT = 2 .AND. ISPRINTER() // ฟงกชันตรวจการมีอยูของเครื่องพิมพ
         SET DEVICE TO PRINTER
      ENDIF
      SET FILTER TO PROREST < REORDER  * 1.1
      GO TOP
      J = 3
      FOR I = 1 TO CNTREORDER
         @ J,2 SAY STR(I) + " | "
         @ J,15 SAY STR(PRO) + " | "
         @ J,25 SAY LEFT(PRONAME,8) + " | "
         @ J,35 SAY STR(PROREST) + " | "
         @ J,45 SAY STR(PROONORD) + " | "
         @ J,55 SAY STR(REORDER) + " | "
         SKIP
         J++
         IF J > 22
            INKEY(0) ; CLS
            J = 3
            IF GASKPRT = 2  // ส ําหรับเครื่องพิมพ
               EJECT

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


336                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

            ENDIF
         ENDIF
      NEXT
      IF GASKPRT = 2
         EJECT ; SET DEVICE TO SCREEN
      ENDIF
   ELSE
      @ 6,5 SAY "ยงัไมมีสินคาชนิดใดใกล ถึงจุดสั่งซื้อ"
   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.29
&   ผลลัพธ
// พมิพรายการสินคาที่ก ําลังสั่งซื้อเพิ่มปจจุบัน โดยเรียงตามชื่อสินคา
// รายงานเลือกเรคอรดที ่จํ านวนสั่งเพิ่มมากกวา 0
// รายงานแสดงล ําดับ รหัสสินคา จํ านวนเหลือ จํ านวนสั่งเพิ่ม จุดสั่งซื้อ
PROCEDURE S0406
   USE PRO
   SORT ON PRONAME TO TEMPF
   USE TEMPF
   COPY TO PRO
   USE PRO
   I = 1
   DO WHILE !EOF()
      IF PROONORD > 0
         ? I,PRO,PROREST,PROONORD,REORDER
         I++
      ENDIF
      SKIP
   ENDDO
   INKEY(3)
   CLOSE
RETURN

: ตัวอยางที่ 6.30
&   ผลลัพธ
// พมิพรายการสินคาที่ขายดีจากมากไปนอย ตามชวงวันที่
// โดยระบุชวงวันเริ่มตน และวันสิ้นสุด ของวันสั่งสินคา
// รายงานแสดงเลขที่ใบสั่ง ปริมาณขาย และวันที่สั่ง
// โดยขอมูลถูกจัดเรียงตามปริมาณขาย จากมากไปนอย
PROCEDURE S0407
   USE ORDERD
   SORT ON OQUANT/D TO TEMPF
   USE TEMPF
   COPY TO ORDERD
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDER
            INDEX ON ORDER TO ORDER1
   SELE 3 ; USE ORDERD
            SET RELATION TO ORDER INTO ORDER,PRO INTO PRO
   _DATEORD1 := _DATEORD2 := CTOD('')
   @ 5,5 SAY "วนัที่สั่งสินคาเริ่มตน = " GET _DATEORD1
   @ 6,5 SAY "วนัที่สั่งสินคาสิ้นสุด = " GET _DATEORD2
   READ
   SET FILTER TO ORDER->DATEORD >= _DATEORD1 .AND. ;
                 ORDER->DATEORD <= _DATEORD2
   GO TOP
   _PROO = 0
   DO WHILE !EOF()
      IF PRO != _PROO
         ? "*  ",PRO,LEFT(PRO->PRONAME,10)
         _PROO = PRO
      ENDIF
      ? ORDER,OQUANT,ORDER->DATEORD
      SKIP
   ENDDO
   INKEY(3)   ;   CLOSE ALL
RETURN

6.4.6 โปรแกรมตาง ๆ ใน SALE5.PRG
โปรแกรมใน SALE5.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 7 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

อืน่ เชน โปรแกรมพิมพสรุปคาขนสงใหบริษัทขนสง  โปรแกรมพิมพจดหมายขอโทษที่สงสินคาลาชากวาวันที่ลูกคา
ตองการสนิคา  โปรแกรมพิมพขอมูลจากแฟมอยางงาย ๆ   โปรแกรมพิมพรายชื่อลูกคาที่พนักงานรับผิดชอบ

ระบบอืน่ เปนระบบที่รวมงานตาง ๆ ที่ไมเขากับระบบยอยอื่น ๆ ซึ่งเปนงานที่หลายหลาย  หากตองการรวม
งานในระบบนีเ้ขากับระบบรายงานก็ยอมท ําได เพราะโปรแกรมเหลานี้เนนการจัดทํ ารายงาน แตผูเขียนไมรวมเขาไป
เพราะงานในระบบรายงานมีมาก  จึงแยกโปรแกรมบางอยางมาสรางระบบใหมจนไดระบบอื่นขึ้นมา

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE5.PRG ทั้งหมด 7 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.31 ถึง ตัวอยางที่ 6.37 ( R0501 - R0507 )
: ตัวอยางที่ 6.31
&   ผลลัพธ
// พมิพสรุปคาขนสงใหบริษัทขนสง โดยระบุเดือน
// รายงานแสดงรหัสบริษัทขนสง ชื่อบริษัทขนสง และจ ํานวนเงินจากการขนสง
PROCEDURE S0501
   _MONTH = 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   READ
   SELE 1 ; USE SHIPPER
            INDEX ON SHIPPER TO SHIPPER1
   SELE 2 ; USE ORDER
            INDEX ON SHIPPER TO ORDER2
            SET RELATION TO SHIPPER INTO SHIPPER
   SET FILTER TO MONTH(DATEORD) = _MONTH
   GO TOP
   _SHIPPERO = 0   ;   _TFREIGHT = 0
   DO WHILE !EOF()
      IF SHIPPER != _SHIPPERO
         ? "========================="
         ? SHIPPER,SHIPPER->SHIPNAME
         _SHIPPERO = SHIPPER
      ENDIF
      ? ORDER,DATEORD,FREIGHT
      _TFREIGHT = _TFREIGHT + FREIGHT

      SKIP
      IF EOF() .OR. _SHIPPERO != SHIPPER
         ? "*" , _TFREIGHT
         _TFREIGHT = 0
      ENDIF
   ENDDO
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.32
&   ผลลัพธ
// พมิพสถติกิารใชบริการบริษัทขนสงของลูกคาแตละราย โดยระบุชวงวันที่
// แยกกลุมขอมูลเปน 2 ระดับ คือแยกรหัสลูกคา แลวจึงแยกบริษัทขนสง
// รายงานแสดงจ ํานวนเงินที่ลูกคาแตละราย ใชบริการแตละบริษัทเทาใด
// สดุทายของลกูคาแตละราย แสดงจํ านวนเงินที่ลูกคาตองจายรวมเทาใด
PROCEDURE S0502
   _DATEORD1 := _DATEORD2 := CTOD('')
   @ 5,5 SAY "วนัที่สั่งสินคาเริ่มตน = " GET _DATEORD1
   @ 6,5 SAY "วนัที่สั่งสินคาสิ้นสุด = " GET _DATEORD2
   READ
   USE ORDER
   INDEX ON STR(CUST)+STR(SHIPPER) TO ORDER3
   SET FILTER TO ORDER->DATEORD >= _DATEORD1 .AND. ;
                 ORDER->DATEORD <= _DATEORD2
   GO TOP
   _CUSTO := _SHIPPERO := 0
   _TFREIGHTC := _TFREIGHTS := 0
   DO WHILE !EOF()
      IF _CUSTO != CUST
         ? "***",CUST
         _CUSTO = CUST
         _SHIPPERO = 0
      ENDIF
      IF _SHIPPERO != SHIPPER

         ? "  *",SHIPPER
         _SHIPPERO = SHIPPER
      ENDIF
      _TFREIGHTC = _TFREIGHTC + FREIGHT
      _TFREIGHTS = _TFREIGHTS + FREIGHT
      SKIP
      IF EOF() .OR. _SHIPPERO != SHIPPER
         ?? _TFREIGHTS
         _TFREIGHTS := 0
      ENDIF
      IF EOF() .OR. _CUSTO != CUST
         ? "  TOTAL = ",_TFREIGHTC
         _TFREIGHTC := 0
      ENDIF
   ENDDO
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.33
&   ผลลัพธ
// พมิพจดหมายขอโทษที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสงสินคา
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE S0503
   _ORDER = 0
   @ 4,5 SAY "ระบุเลขที่ใบสงสินคา : " GET _ORDER PICT "99999"
   READ
   IF LASTKEY() != 27
      GASKPRT = ASKPRT()
      IF GASKPRT = 2
         SET DEVICE TO PRINTER
      ENDIF
      @ 5,5 SAY "เรียน ทานผูมีอุปการะคุณ"
      @ 6,5 SAY "ตามที่ทานไดสั่งสินคากับบริษัทของเรา"
      @ 7,5 SAY "โดยมรีายการตามใบสั่งสินคาเลขที ่"+STR(_ORDER)

      @ 8,5 SAY "แลวทางบริษัทของเราไดสงสินคาใหทานชากวากํ าหนด"
      @ 9,5 SAY "กระผมในฐานะประธานกรรมการของบริษัท"
      @ 10,5 SAY "ใครขออภัยเปนอยางสูง และไดช ําระคาปรับมาพรอมกันนี้"
      @ 11,5 SAY "ทายนี ้กระผมขอใหคํ ามั่นวาเราจะปรับปรุงระบบจัดสง"
      @ 12,5 SAY "ใหเรว็ทันก ําหนดเวลาที่ทานตองการทุกครั้งไป"
      @ 13,5 SAY "   จงึเรียนมาเพื่อกราบขออภัยมา ณ ที่นี"้
      @ 14,5 SAY "                      ประธานบริษัทขายสง จํ ากัด"
      IF GASKPRT = 2
         EJECT
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(3)
   ENDIF
RETURN
: ตัวอยางที่ 6.34
&   ผลลัพธ
// พมิพชื่อลูกคาที่พนักงานแตละคนรับผิดชอบ โดยระบุชวงวันที่
// รายงานแยกกลุมขอมูลตามรหัสพนักงาน
// รายงานแสดงรหัสและชื่อพนักงาน รหัสลูกคา ชื่อลูกคา เลขที่ใบสั่ง วันที่สั่ง
PROCEDURE S0504
   _DATEORD1 := _DATEORD2 := CTOD('')
   @ 5,5 SAY "วนัที่สั่งสินคาเริ่มตน = " GET _DATEORD1
   @ 6,5 SAY "วนัที่สั่งสินคาสิ้นสุด = " GET _DATEORD2
   READ
   IF LASTKEY() != 27
      SELE 1 ; USE EMPL INDEX EMPL1
      SELE 2 ; USE CUST
               INDEX ON CUST TO CUST1
      SELE 3 ; USE ORDER
               INDEX ON EMPL TO ORDER4
               SET RELATION TO CUST INTO CUST,EMPL INTO EMPL
               SET FILTER TO DATEORD >= _DATEORD1 .AND. ;
                             DATEORD <= _DATEORD2
      GO TOP

      _EMPLO = 0
      DO WHILE !EOF()
         IF EMPL != _EMPLO
            ? "*",EMPL,EMPL->EMPLNAME
            _EMPLO = EMPL
         ENDIF
         ? "-",CUST,CUST->CUSTNAME,ORDER,DATEORD
         SKIP
      ENDDO
      INKEY(3)
      CLOSE ALL
   ENDIF
RETURN
: ตัวอยางที่ 6.35
&   ผลลัพธ
// พมิพเลขที่ใบสั่งสินคาที่สงไมทันก ําหนด โดยระบุเดือน
// เลอืกเดือนที่ตรงกับวันสั่ง วันขนสง หรือวันที่ลูกคาตองการ
// รายงานแสดงขอมูลทุกฟลดจากแฟมใบสั่งสินคา
PROCEDURE S0505
   _MONTH := 0
   @ 5,5 SAY "เดอืนที่ตองการรายงาน = " GET _MONTH PICT "99"
   READ
   IF LASTKEY() != 27
      USE ORDER
      SET FILTER TO (MONTH(DATEORD) = _MONTH .OR. ;
                    MONTH(DATEREQ) = _MONTH .OR. ;
                    MONTH(DATESHIP) = _MONTH) .AND. ;
                    DATESHIP > DATEREQ
      COUNT TO CNTREC
      IF CNTREC > 0
         GO TOP
         DO WHILE !EOF()
            ? ORDER,CUST,EMPL,DATEORD,DATEREQ,DATESHIP,SHIPPER,FREIGHT
            SKIP

         ENDDO
      ELSE
         @ 6,5 SAY "ไมพบขอมูลที่ถูกสั่งในเดือนนี้เลย"
      ENDIF
      INKEY(3)
      CLOSE ALL
   ENDIF
RETURN
: ตัวอยางที่ 6.36
&   ผลลัพธ
// พมิพเลขที่ใบสั่งสินคาที่มีก ําหนดสงไมเกิน 10 วันนับจากวันนี้
// รายงานแสดงขอมูลทุกฟลด จากแฟมใบสั่งซื้อ
PROCEDURE S0506
   USE ORDER
   SET FILTER TO DATESHIP < DATE() + 10 .AND. ;
                 DATESHIP > DATE()        // วนัที่สงตองยังไมถึง
   COUNT TO  CNTREC
   IF CNTREC > 0
      GO TOP
      DO WHILE !EOF()
         ? ORDER,CUST,EMPL,DATEORD,DATEREQ,DATESHIP,SHIPPER,FREIGHT
         SKIP
      ENDDO
   ELSE
      ? "ไมมรีายการที่ตองจัดสงใน 10 วันขางหนา"
   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.37
&   ผลลัพธ
// พมิพขอมูลจากแฟมตามที่เลือก โดยเลือกแฟมจากตัวเลือก แลวเปดแฟม
// น ําขอมูลในแฟมมาพิมพ โดยใชฟงกชันอานชื่อฟลดอัตโนมัติ
// สามารถเลือกปลายทางของผลลัพธได

PROCEDURE S0507
   OMENU = ALERTNEW("เลอืกปรับปรุงแฟมตามตองการ",;
          {"แฟมสั่งสินคา","แฟมสั่งสินคาละเอียด","แฟมสินคา",;
           "แฟมพนักงาน","แฟมลูกคา","แฟมบริษัทรับสงของ",;
           "แฟมประเภทสินคา","แฟมผูผลิต","กลับเมนู"})
   IF MOD(OMENU,9) != 0
      GASKPRT = ASKPRT()
      IF !ISPRINTER()
         GASKPRT = 1
         @ 2,5 SAY "บอกใหทราบวา ... เครื่องพิมพปดอยู"
      ENDIF
      MAXLINE = 20
      IF GASKPRT = 2
         SET DEVICE TO PRINTER
         MAXLINE = 35
         ? "รายงานผลทางเครื่องพิมพ ... โปรดรอสักครู"
      ENDIF
      DO CASE
         CASE OMENU = 1 ; USE ORDER
         CASE OMENU = 2 ; USE ORDERD
         CASE OMENU = 3 ; USE PRO
         CASE OMENU = 4 ; USE EMPL
         CASE OMENU = 5 ; USE CUST
         CASE OMENU = 6 ; USE SHIPPER
         CASE OMENU = 7 ; USE CAT
         CASE OMENU = 8 ; USE SUP
      ENDCASE
      ARFLDNAME = ARRAY(AFIELDS())   // สรางอาเรยเทาจํ านวนฟลดของแฟม
      ARFLDTYPE = ARRAY(AFIELDS())   // สรางอาเรยเทาจํ านวนฟลดของแฟม
      ARFLDWIDTH= ARRAY(AFIELDS())   // สรางอาเรยเทาจํ านวนฟลดของแฟม
      AFIELDS(ARFLDNAME,ARFLDTYPE,ARFLDWIDTH)  // เก็บชื่อฟลด
      LN := 1 ; NUM =  1
      DO WHILE !EOF()
         @ LN+2,2 SAY LTRIM(STR(NUM,3)) + " : "
         FOR I = 1 TO AFIELDS()
             IF GASKPRT = 2

                @ LN+2,PCOL() SAY &(ARFLDNAME[I])
                @ LN+2,PCOL() SAY " | "
             ELSE
                @ LN+2,COL() SAY &(ARFLDNAME[I])
                @ LN+2,COL() SAY " | "
             ENDIF
         NEXT
         IF MOD(LN,MAXLINE) = 0 .AND. LN != 0
            IF GASKPRT = 2
               EJECT
            ENDIF
            ? "กดปุมใด ๆ เพื่อพิมพหนาตอไป"
            LN = 0  ; INKEY(3) ; CLS
         ENDIF
         LN++ ; NUM++
         SKIP
      ENDDO
      IF GASKPRT = 2
         EJECT
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(5)
   ENDIF
   CLOSE ALL
RETURN

